

# SYSTÈMES D'ÉQUATIONS DIFFÉRENTIELLES : EXERCICES CORRIGÉS

Bernard Dupont

[Bernard.Dupont@univ-lille1.fr](mailto:Bernard.Dupont@univ-lille1.fr)

## Exercice M1

### Enoncé

Résoudre explicitement les systèmes de deux équations différentielles suivants :

1.

$$\begin{cases} x'(t) = x(t) + y(t) \\ y'(t) = 2x(t) \end{cases}$$

2.

$$\begin{cases} x'(t) = 2x(t) - 2y(t) \\ y'(t) = x(t) - y(t) \end{cases}$$

### Solution

Cet exercice ne présente aucune difficulté, d'autant plus que les systèmes à étudier sont linéaires homogènes. La commande **dsolve** n'est pas perturbée par le fait que le second système admet une infinité d'équilibres.

```
> dsolve({diff(x(t),t)=x(t)+y(t),diff(y(t),t)=2*x(t)},{x(t),y(t)});
```

$$\left\{ x(t) = -\frac{1}{2} \_C1 e^{-t} + \_C2 e^{2t}, y(t) = \_C1 e^{-t} + \_C2 e^{2t} \right\} \quad (1.1)$$

```
> dsolve({diff(x(t),t)=2*x(t)-2*y(t),diff(y(t),t)=x(t)-y(t)},{x(t),y(t)});
```

$$\left\{ x(t) = \_C1 + \_C2 e^t, y(t) = \frac{1}{2} \_C2 e^t + \_C1 \right\} \quad (1.2)$$

## Exercice M2

### Enoncé

On considère le système de deux équations différentielles suivant :

$$\begin{cases} x'(t) = x(t) + y(t) \\ y'(t) = 2x(t) - y(t) \end{cases}$$

Résoudre explicitement ce système en tenant compte des conditions suivantes :

1.  $x(0) = y(0) = 1$

2.

$$x(1) = \frac{1}{\sqrt{3}} \text{ et } y(5) = \frac{1}{5\sqrt{3}}$$

3.  $x(0) = 1$  et  $x'(0) = 1$

4.  $x'(0) = -1$  et  $y'(0) = -1$

Dans chaque cas, représenter graphiquement les trajectoires des fonctions solutions.


### Solution

Il s'agit de résoudre des problèmes de Cauchy avec diverses conditions "initiales" prouvant que de telles conditions ne portent pas systématiquement sur les valeurs prises par les variables d'état à la date  $t = 0$ .

1.

```
> restart;
dsolve({diff(x(t),t)=x(t)+y(t),diff(y(t),t)=2*x(t)-y(t),x(0)=
1,y(0)=1},{x(t),y(t)});
assign(%);
plot({x(t),y(t)},t=0..2);
```

$$\left\{ x(t) = \frac{1}{2} \left( \frac{1}{6} \sqrt{3} + \frac{1}{2} \right) \sqrt{3} e^{\sqrt{3}t} - \frac{1}{2} \left( -\frac{1}{6} \sqrt{3} + \frac{1}{2} \right) \sqrt{3} e^{-\sqrt{3}t} + \frac{1}{2} \left( \frac{1}{6} \sqrt{3} + \frac{1}{2} \right) e^{\sqrt{3}t} + \frac{1}{2} \left( -\frac{1}{6} \sqrt{3} + \frac{1}{2} \right) e^{-\sqrt{3}t}, y(t) = \left( \frac{1}{6} \sqrt{3} + \frac{1}{2} \right) e^{\sqrt{3}t} + \left( -\frac{1}{6} \sqrt{3} + \frac{1}{2} \right) e^{-\sqrt{3}t} \right\}$$




2. La condition porte sur la variable  $x$  à la date 1 et la variable  $y$  à la date 5.

```
> restart;
dsolve({diff(x(t),t)=x(t)+y(t),diff(y(t),t)=2*x(t)-y(t),x(1)=
1/sqrt(3),y(5)=1/(5*sqrt(3))},{x(t),y(t)});
assign(%);
plot({x(t),y(t)},t=0..7);
```

$$\left\{ x(t) = \frac{1}{30} \frac{(\sqrt{3} + 3) (5 + 5\sqrt{3} + (e^{\sqrt{3}})^4) e^{\sqrt{3}t}}{e^{\sqrt{3}t} (\sqrt{3} + 2 + (e^{\sqrt{3}})^8)} \right\}$$

$$\begin{aligned}
& - \frac{1}{30} \frac{\sqrt{3} (e^{\sqrt{3}})^5 (-10 (e^{\sqrt{3}})^4 + 1 + \sqrt{3}) e^{-\sqrt{3} t}}{\sqrt{3} + 2 + (e^{\sqrt{3}})^8}, y(t) \\
& = \frac{1}{30} \frac{(\sqrt{3} + 3) (5 + 5\sqrt{3} + (e^{\sqrt{3}})^4) \sqrt{3} e^{\sqrt{3} t}}{e^{\sqrt{3}} (\sqrt{3} + 2 + (e^{\sqrt{3}})^8)} \\
& + \frac{1}{10} \frac{(e^{\sqrt{3}})^5 (-10 (e^{\sqrt{3}})^4 + 1 + \sqrt{3}) e^{-\sqrt{3} t}}{\sqrt{3} + 2 + (e^{\sqrt{3}})^8} \\
& - \frac{1}{30} \frac{(\sqrt{3} + 3) (5 + 5\sqrt{3} + (e^{\sqrt{3}})^4) e^{\sqrt{3} t}}{e^{\sqrt{3}} (\sqrt{3} + 2 + (e^{\sqrt{3}})^8)} \\
& + \frac{1}{30} \frac{\sqrt{3} (e^{\sqrt{3}})^5 (-10 (e^{\sqrt{3}})^4 + 1 + \sqrt{3}) e^{-\sqrt{3} t}}{\sqrt{3} + 2 + (e^{\sqrt{3}})^8} \}
\end{aligned}$$




3. Les conditions portent sur la variable  $x$  et sa dérivée à la date  $t = 0$ .

```

> restart;
dsolve({diff(x(t),t)=x(t)+y(t),diff(y(t),t)=2*x(t)-y(t),x(0)=
1,D(x)(0)=1},{x(t),y(t)});
assign(%);
plot({x(t),y(t)},t=0..3);

```


$$\left\{ x(t) = \frac{1}{2} e^{\sqrt{3} t} + \frac{1}{2} e^{-\sqrt{3} t} + \frac{1}{6} \sqrt{3} e^{\sqrt{3} t} - \frac{1}{6} \sqrt{3} e^{-\sqrt{3} t}, y(t) = \frac{1}{3} \sqrt{3} e^{\sqrt{3} t} - \frac{1}{3} \sqrt{3} e^{-\sqrt{3} t} \right\}$$



4. Les conditions portent sur les dérivées des variables  $x$  et  $y$  à la date  $t = 0$ .

```
> restart;
dsolve({diff(x(t),t)=x(t)+y(t),diff(y(t),t)=2*x(t)-y(t),D(x)
(0)=-1,D(y)(0)=-1},{x(t),y(t)});
assign(%);
plot({x(t),y(t)},t=0..3);
```

$$\left\{ \begin{aligned} x(t) &= \left( -\frac{1}{6}\sqrt{3} - \frac{1}{3} \right) e^{\sqrt{3}t} + \left( \frac{1}{6}\sqrt{3} - \frac{1}{3} \right) e^{-\sqrt{3}t}, y(t) = \left( -\frac{1}{6}\sqrt{3} \right. \\ &\quad \left. - \frac{1}{3} \right) \sqrt{3} e^{\sqrt{3}t} - \left( \frac{1}{6}\sqrt{3} - \frac{1}{3} \right) \sqrt{3} e^{-\sqrt{3}t} - \left( -\frac{1}{6}\sqrt{3} - \frac{1}{3} \right) e^{\sqrt{3}t} - \left( \frac{1}{6}\sqrt{3} \right. \\ &\quad \left. - \frac{1}{3} \right) e^{-\sqrt{3}t} \end{aligned} \right\}$$



## ▼ Exercice M3

[Enoncé

1. Résoudre le système de deux équations différentielles suivant :

$$\begin{cases} x''(t) + 3y'(t) - 4x(t) + 6y(t) = 0 \\ x'(t) + y''(t) - 2x(t) + 4y(t) = 0 \end{cases}$$

2. Montrer que le profil de la trajectoire de  $x$  et l'orbite  $(x(t), y(t))$  dépendent des valeurs prises par les paramètres  $_C1$ ,  $_C2$ ,  $_C3$  et  $_C4$ .

### Solution

1. Il s'agit d'un système de deux équations linéaires homogènes du second ordre (les dérivées secondes des fonctions  $x$  et  $y$  sont présentes) qui ne dérouté pas l'instruction **dsolve** :

```
> restart;
```

$$\text{soleqd} := \text{dsolve}(\{\text{diff}(x(t), t, t) + 3 * \text{diff}(y(t), t) - 4 * x(t) + 6 * y(t) = 0, \text{diff}(x(t), t) + \text{diff}(y(t), t, t) - 2 * x(t) + 4 * y(t) = 0\}, \{x(t), y(t)\});$$

$$\text{soleqd} := \left\{ x(t) = \_C1 e^{-2t} + \_C2 e^{2t} + \_C3 \sin(t) + \_C4 \cos(t), y(t) = \frac{1}{2} \_C1 e^{-2t} - \frac{1}{3} \_C3 \cos(t) + \frac{1}{3} \_C4 \sin(t) + \frac{2}{3} \_C3 \sin(t) + \frac{2}{3} \_C4 \cos(t) \right\} \quad (3.1)$$

Pour préparer la question suivante, on récupère les solutions en  $x$  et  $y$  comme des fonctions de  $t$  et des paramètres générés automatiquement par Maple.

```
> assign(soleqd);
```

$$X := \text{unapply}(x(t), (t, \_C1, \_C2, \_C3, \_C4));$$

$$Y := \text{unapply}(y(t), (t, \_C1, \_C2, \_C3, \_C4));$$


$$X := (t, \_C1, \_C2, \_C3, \_C4) \rightarrow \_C1 e^{-2t} + \_C2 e^{2t} + \_C3 \sin(t) + \_C4 \cos(t)$$

$$Y := (t, \_C1, \_C2, \_C3, \_C4) \rightarrow \frac{1}{2} \_C1 e^{-2t} - \frac{1}{3} \_C3 \cos(t) + \frac{1}{3} \_C4 \sin(t) + \frac{2}{3} \_C3 \sin(t) + \frac{2}{3} \_C4 \cos(t) \quad (3.2)$$

2. Au vu des solutions, la trajectoire de  $y$  est oscillatoire alors que celle de  $x$  est explosive si  $_C2 \neq 0$  et oscillatoire sinon.


Commençons par représenter le cas où les 2 trajectoires oscillent perpétuellement. On prend  $_C2 = 0$ .

```
> plot([X(t, 1, 0, 1, 1), Y(t, 1, 0, 1, 1)], t=0..25, color=[black, red], thickness=2, legend=[typeset('X(t)'), typeset('Y(t)')]);
```




L'orbite correspondante devient rapidement une ellipse de centre O.

```
> plot([X(t,1,0,1,1),Y(t,1,0,1,1)],t=0..25,x=-2..2,y=-2..2);
```




Pour  $C_2 \neq 0$ , la trajectoire de  $y$  reste oscillatoire mais celle de  $x$  devient explosive. Par exemple, pour  $C_2 = 10^{-3}$ , on obtient les trajectoires :

```
> plot([X(t,1,0.001,1,1),Y(t,1,0.001,1,1)],t=0..10,y=-5..20,
color=[black,red],thickness=2,legend=[typeset('X(t)'),typeset('Y(t)')]);
```



et l'orbite :

```
> plot([X(t,1,0.001,1,1),Y(t,1,0.001,1,1)],t=0..10],x=-2..20,y=-2..2);
```



## Exercice M4

### Enoncé

Résoudre le système différentiel suivant :

$$\begin{cases} x' = x - y + 4z \\ y' = 3x + 2y - z \\ z' = 2x + y - z \end{cases}$$

On utilisera la notation en **D** pour écrire les équations différentielles dans **dplot**.

### Solution

Il s'agit d'un système de trois équations différentielles linéaires homogènes du premier ordre. L'énoncé invite à utiliser l'opérateur de dérivation **D** plutôt que **diff**, ce qui fait que les inconnues sont les fonctions **x**, **y** et **z** (et non les expressions **x(t)**, **y(t)** et **z(t)**)

```
> restart;
dsolve({D(x)(t)=x(t)-y(t)+4*z(t),D(y)(t)=3*x(t)+2*y(t)-z(t),D
(z)(t)=2*x(t)+y(t)-z(t)},{x,y,z});
{x(t)=-_C1 e^t+_C2 e^{3t}+_C3 e^{-2t},y(t)=-4 _C1 e^t+2 _C2 e^{3t}-_C3 e^{-2t},z(t)=
-_C1 e^t+_C2 e^{3t}-_C3 e^{-2t}}
```

(4.1)

## Exercice M5

### Enoncé

Caractériser, résoudre à la main puis résoudre avec Maple les systèmes suivants :

$$1. \begin{cases} x'(t) + y'(t) - y(t) = \sin(t) \\ x'(t) - y'(t) - x(t) = t \end{cases}$$

$$2. \begin{cases} 2x'(t) + y'(t) - 3x(t) - y(t) = t \\ x'(t) + y'(t) - 4x(t) - y(t) = e^t \end{cases}$$

$$3. \begin{cases} x'(t) = \frac{1}{t}x(t) - \frac{1}{t^2}y(t) \\ y'(t) = 2x(t) - \frac{1}{t}y(t) \end{cases}$$

### Solution

1. Même s'il est déguisé, il s'agit d'un système de deux équations différentielles linéaires du premier ordre à coefficients constants et avec second membre. La vitesse d'exécution de Maple est stupéfiante ...

```
> restart;
dsolve({diff(x(t),t)+diff(y(t),t)-y(t)=sin(t),diff(x(t),t)-
diff(y(t),t)-x(t)=t},{x(t),y(t)});
{x(t)=-e^{\frac{1}{2}t} \cos\left(\frac{1}{2}t\right) _C2 + e^{\frac{1}{2}t} \sin\left(\frac{1}{2}t\right) _C1 - \frac{1}{5} \cos(t) - \frac{2}{5} \sin(t) - t - 1, y(t)
=e^{\frac{1}{2}t} \sin\left(\frac{1}{2}t\right) _C2 + e^{\frac{1}{2}t} \cos\left(\frac{1}{2}t\right) _C1 - \frac{1}{5} \sin(t) - \frac{3}{5} \cos(t) - 1}
```

(5.1)

2. Ce second système a les mêmes caractéristiques que le précédent :

```
> restart;
dsolve({2*diff(x(t),t)+diff(y(t),t)-3*x(t)-y(t)=t,diff(x(t),
t)+diff(y(t),t)-4*x(t)-y(t)=exp(t)},{x(t),y(t)});
```

(5.2)

$$\left\{ x(t) = t - 1 - \frac{1}{2} e^t + e^{-t} \_C2, y(t) = -4t + 1 - \frac{1}{2} e^t t - \frac{5}{2} e^{-t} \_C2 + e^t \_C1 \right\} \quad (5.2)$$

3. Il s'agit d'un système de deux équations différentielles linéaires du premier ordre à coefficients variables et sans second membre. Les solutions sont définies sur tout intervalle ne contenant pas 0.

> restart;

```
dsolve({diff(x(t),t)=(1/t)*x(t)-(1/t^2)*y(t),diff(y(t),t)=2*x(t)-(1/t)*y(t)},{x(t),y(t)});
```

$$\left\{ x(t) = \_C1 + \frac{\_C2}{t}, y(t) = 2 \_C2 + \_C1 t \right\} \quad (5.3)$$

## Exercice M6

### Enoncé

On considère le système différentiel suivant :

$$\begin{cases} x' = -x + y \\ y' = -y + z \\ z' = -z + e^{-t} \end{cases}$$

Ecrire ce système sous forme matricielle. Quelle est sa particularité?

Intégrer ce système avec **dsolve** puis représenter, à l'aide de deux méthodes différentes, l'orbite démarrant en  $x(0) = y(0) = z(0) = 0$ .

### Solution

Sous la forme matricielle  $X' = AX + B$  où  $X = \begin{pmatrix} x \\ y \\ z \end{pmatrix}$ ,  $A = \begin{pmatrix} -1 & 1 & 0 \\ 0 & -1 & 1 \\ 0 & 0 & -1 \end{pmatrix}$  et  $B = \begin{pmatrix} 0 \\ 0 \\ e^{-t} \end{pmatrix}$ , on

constate que la matrice  $A$  est trigonale (-1 est valeur propre d'ordre 3). On peut résoudre ce système "en cascade" en commençant par la 3<sup>e</sup> équation, puis passer à la 2<sup>e</sup>, enfin résoudre la 1<sup>e</sup>. Des erreurs de calcul sont toujours possibles. Autant se fier aux automatismes de Maple.

> restart;

```
eqd:={diff(x(t),t)=-x(t)+y(t),diff(y(t),t)=-y(t)+z(t),diff(z(t),t)=-z(t)+exp(-t)};
```

```
inc:={x(t),y(t),z(t)};
```

```
dsolve(eqd,inc);
```

$$eqd := \left\{ \frac{d}{dt} x(t) = -x(t) + y(t), \frac{d}{dt} y(t) = -y(t) + z(t), \frac{d}{dt} z(t) = -z(t) + e^{-t} \right\}$$

$$inc := \{x(t), y(t), z(t)\}$$

$$\left\{ x(t) = \frac{1}{6} (6t \_C2 + 3t^2 \_C3 + t^3 + 6 \_C1) e^{-t}, y(t) = \frac{1}{2} (2t \_C3 + t^2 + 2 \_C2) e^{-t}, \right. \quad (6.1)$$

$$\left. z(t) = (t + \_C3) e^{-t} \right\}$$

Pour les conditions initiales précisées, on obtient la solution :

> condinit:={x(0)=0,y(0)=0,z(0)=0};

```
soleqd:=dsolve(eqd union condinit,inc);
```


```
assign(soleqd);
```

```
condinit := {x(0) = 0, y(0) = 0, z(0) = 0}
```

$$\text{soleqd} := \left\{ x(t) = \frac{1}{6} e^{-t} t^3, y(t) = \frac{1}{2} e^{-t} t^2, z(t) = e^{-t} t \right\} \quad (6.2)$$


L'orbite correspondante peut être tracée avec `plot3d` :

```
> plot3d([x(t),y(t),z(t)],t=0..10,z=0..25,axes=boxed,thickness=
2,color=red,grid=[50,50],orientation=[55,71]);
```



ou, alternativement, avec la commande `DEplot3d` du paquetage `DEtools` (noter que les fonctions sont renommées puisque les fonctions  $x(t)$ ,  $y(t)$  et  $z(t)$  ont été assignées).

```
> with(DEtools):
DEplot3d({diff(a(t),t)=-a(t)+b(t),diff(b(t),t)=-b(t)+c(t),
diff(c(t),t)=-c(t)+exp(-t)},[a(t),b(t),c(t)],t=0..10,[[a(0)=
0,b(0)=0,c(0)=0]],stepsize=0.01,linecolor=black,orientation=
[55,71]);
```



## Exercice M7

### Enoncé

Soit le système différentiel  $(S)$  suivant :

$$\begin{cases} x' = x + y - z + e^t \\ y' = -x + y + z \\ z' = x - y - z \end{cases}$$

1. Ecrire  $(S)$  sous la forme matricielle  $X' = AX + B$  puis vérifier que deux valeurs propres de  $A$  sont complexes.
2. Déterminer les solutions réelles de  $(S)$ .
3. Représenter trois orbites solutions de  $(S)$ .

### Solution

1. On a  $X' = AX + B$  où  $X = \begin{pmatrix} x \\ y \\ z \end{pmatrix}$ ,  $A = \begin{pmatrix} 1 & 1 & -1 \\ -1 & 1 & 1 \\ 1 & -1 & -1 \end{pmatrix}$  et  $B = \begin{pmatrix} e^t \\ 0 \\ 0 \end{pmatrix}$ .

Avec la commande **Eigenvalues** du paquetage **LinearAlgebra**, on vérifie que  $A$  admet deux valeurs propres complexes conjuguées.

```
> restart;
A:=Matrix(3,3,[[1,1,-1],[-1,1,1],[1,-1,1]]);
with(LinearAlgebra):
Eigenvalues(A);
```

$$A := \begin{bmatrix} 1 & 1 & -1 \\ -1 & 1 & 1 \\ 1 & -1 & 1 \end{bmatrix} \quad \begin{bmatrix} 1 \\ 1 + I\sqrt{3} \\ 1 - I\sqrt{3} \end{bmatrix} \quad (7.1)$$

En soi, cette situation n'est pas dramatique. Mais s'il faut conduire tous les calculs de recherche des vecteurs propres, d'inversion de la matrice de passage, de résolution dans la base de vecteurs propres, etc. à la main, le risque d'erreur est diablement élevé. D'où le recours à **dsolve** dans la question suivante.

2. Maple affiche directement les solutions réelles de (S) ...

```
> eqd:=diff(x(t),t)=x(t)+y(t)-z(t)+exp(t),diff(y(t),t)=-x(t)+y(t)+z(t),diff(z(t),t)=x(t)-y(t)+z(t);
inc:=x(t),y(t),z(t);
soleqd:=dsolve({eqd},{inc});
```

$$eqd := \frac{d}{dt} x(t) = x(t) + y(t) - z(t) + e^t, \frac{d}{dt} y(t) = -x(t) + y(t) + z(t), \frac{d}{dt} z(t) = x(t) - y(t) + z(t)$$

$$inc := x(t), y(t), z(t)$$


$$soleqd := \left\{ x(t) = -\frac{1}{6} e^t (2 - 2t - 6\_C1 + 3\_C2 \cos(\sqrt{3} t) + 3\_C2 \sin(\sqrt{3} t) \sqrt{3} + 3\_C3 \sin(\sqrt{3} t) - 3\_C3 \cos(\sqrt{3} t) \sqrt{3}), y(t) = \frac{1}{6} e^t (-4 + 2t + 6\_C1 - 3\_C2 \cos(\sqrt{3} t) + 3\_C2 \sin(\sqrt{3} t) \sqrt{3} - 3\_C3 \sin(\sqrt{3} t) - 3\_C3 \cos(\sqrt{3} t) \sqrt{3}), z(t) = \frac{1}{3} e^t (t + 3\_C1 + 3\_C2 \cos(\sqrt{3} t) + 3\_C3 \sin(\sqrt{3} t)) \right\} \quad (7.2)$$

3. Les trois orbites peuvent être tracées à l'aide de la commande **DEplot3d** du paquetage **DEtools**.

```
> restart;
with(DEtools):
eqd:=diff(x(t),t)=x(t)+y(t)-z(t)+exp(t),diff(y(t),t)=-x(t)+y(t)+z(t),diff(z(t),t)=x(t)-y(t)+z(t);
inc:=x(t),y(t),z(t);
DEplot3d([eqd],[inc],t=0..5,[[x(0)=0,y(0)=0,z(0)=0],[x(0)=1,y(0)=1,z(0)=1],[x(0)=-1,y(0)=-1,z(0)=-1]],linecolor=[black,red,blue],orientation=[56,85],thickness=2);
```

$$eqd := \frac{d}{dt} \begin{bmatrix} x(t) \\ y(t) \\ z(t) \end{bmatrix} = \begin{bmatrix} x(t) + y(t) - z(t) + e^t \\ -x(t) + y(t) + z(t) \\ x(t) - y(t) + z(t) \end{bmatrix}$$

$$inc := x(t), y(t), z(t)$$



## Exercice M8

### Enoncé

On considère le système différentiel suivant :

$$\begin{cases} x' = -y + \sin(\alpha t) \\ y' = x - \cos(\alpha t) \end{cases}$$

où  $\alpha$  est un réel.

1. Résoudre le système quand  $\alpha = 1$ . Représenter les trajectoires de  $x(t)$  et  $y(t)$  pour les conditions initiales  $(x(0), y(0)) = (0, 0)$ ,  $(x(0), y(0)) = (0, 2)$ ,  $(x(0), y(0)) = (0, 4)$  et  $(x(0), y(0)) = (0, 6)$ . A l'aide de deux méthodes différentes, représenter les quatre orbites correspondantes dans le plan des phases.

2. Même question que précédemment en prenant  $\alpha \neq 1$  et  $\alpha \neq -1$ .

### Solution

Il s'agit d'un système différentiel linéaire à coefficients constants en dimension 2 avec second membre dépendant du temps.

1. Si  $\alpha = 1$ , il y a résonance et les deux trajectoires-solutions sont explosives.

> **restart;**

**eqd:={diff(x(t),t)=-y(t)+sin(t),diff(y(t),t)=x(t)-cos(t)};**

**condinit:={x(0)=x0,y(0)=y0};**

```
inc:={x(t),y(t)};
```

```
soleqd:=dsolve(eqd union condinit,inc);
```

$$eqd := \left\{ \frac{d}{dt} x(t) = -y(t) + \sin(t), \frac{d}{dt} y(t) = x(t) - \cos(t) \right\}$$

$$condinit := \{x(0) = x0, y(0) = y0\}$$

$$inc := \{x(t), y(t)\}$$

$$soleqd := \{x(t) = -\sin(t) y0 + \cos(t) (-1 + x0) + \cos(t) + \sin(t) t, y(t) = \cos(t) y0 + \sin(t) (-1 + x0) - \cos(t) t + \sin(t)\} \quad (8.1)$$

Pour représenter les trajectoires, on extrait les expressions des solutions avec **assign** qu'on transforme aussitôt avec **unapply** en fonctions de trois variables : deux pour les conditions initiales et la troisième pour le temps.

```
> assign(soleqd);
```

```
X:=unapply(x(t),(x0,y0,t));
```

```
Y:=unapply(y(t),(x0,y0,t));
```

$$X := (x0, y0, t) \rightarrow -\sin(t) y0 + \cos(t) (-1 + x0) + \cos(t) + \sin(t) t$$

$$Y := (x0, y0, t) \rightarrow \cos(t) y0 + \sin(t) (-1 + x0) - \cos(t) t + \sin(t) \quad (8.2)$$


Il ne reste plus qu'à tracer les trajectoires avec **plot**.



```
> plot([X(0,0,t),Y(0,0,t)],t=0..4*Pi);
```


```
plot([X(0,2,t),Y(0,2,t)],t=0..4*Pi);
```

```
plot([X(0,4,t),Y(0,4,t)],t=0..4*Pi);
```

```
plot([X(0,6,t),Y(0,6,t)],t=0..4*Pi);
```








La première méthode de mise en place des 4 orbites correspondantes dans le plan des phases consiste à superposer les 4 courbes paramétrées obtenues à partir des fonctions-solutions. On obtient évidemment des "spiraloïdes" explosives.

```
> g1:=plot([X(0,0,t),Y(0,0,t),t=0..20],x=-20..20,y=-20..20,
color=black);
g2:=plot([X(0,2,t),Y(0,2,t),t=0..20],x=-20..20,y=-20..20,
color=blue);
g3:=plot([X(0,4,t),Y(0,4,t),t=0..20],x=-20..20,y=-20..20,
color=red);
g4:=plot([X(0,6,t),Y(0,6,t),t=0..20],x=-20..20,y=-20..20,
color=green);
with(plots):
display(g1,g2,g3,g4);
      g1 := PLOT(...)
      g2 := PLOT(...)
      g3 := PLOT(...)
      g4 := PLOT(...)
```




L'autre méthode fait appel à la commande **DEplot** du paquetage **DEtools**. Avant de lancer la requête, il faut vider le buffer par un **restart** et réécrire le système différentiel ainsi que les inconnues puisque les fonctions-expressions **x(t)** et **y(t)** ont été assignées ci-dessus.

```
> restart;
eqd:={diff(x(t),t)=-y(t)+sin(t),diff(y(t),t)=x(t)-cos(t)};
inc:={x(t),y(t)};
with(DEtools):
DEplot(eqd,inc,t=0..20,[x(0)=0,y(0)=0],[x(0)=0,y(0)=2],[x(0)
=0,y(0)=4],[x(0)=0,y(0)=6]],stepsize=0.01,linecolor=[black,
blue,red,green],thickness=1);
```

$$eqd := \left\{ \frac{d}{dt} x(t) = -y(t) + \sin(t), \frac{d}{dt} y(t) = x(t) - \cos(t) \right\}$$

$$condinit := \{x(0) = x0, y(0) = y0\}$$

$$inc := \{x(t), y(t)\}$$



2. On reprend la méthode d'exploration de la question précédente.

Surprise! Quand on demande la résolution du système différentiel dans le cas général (pour  $\alpha$  quelconque), Maple distingue trois cas de figure :  $\alpha = -1$ ,  $\alpha = 1$  et  $\alpha$  différent de  $-1$  et  $1$ .

```
> restart;
eqd:={diff(x(t),t)=-y(t)+sin(alpha*t),diff(y(t),t)=x(t)-cos(alpha*t)};
condinit:={x(0)=x0,y(0)=y0};
inc:={x(t),y(t)};
soleq:=dsolve(eqd union condinit,inc);
```

$$eqd := \left\{ \frac{d}{dt} x(t) = -y(t) + \sin(\alpha t), \frac{d}{dt} y(t) = x(t) - \cos(\alpha t) \right\}$$

$$condinit := \{x(0) = x_0, y(0) = y_0\}$$

$$inc := \{x(t), y(t)\}$$

$$soleq := \left\{ x(t) = -\sin(t) y_0 + \frac{1}{2} \frac{1}{-1 + \alpha} (\cos(t) (-2 \operatorname{Dirac}(\alpha + 1) - \operatorname{Dirac}(-1 + \alpha)) \right. \quad (8.3)$$

$$\left. + 2 \operatorname{Dirac}(-1 + \alpha) \operatorname{Dirac}(\alpha + 1) - \operatorname{Dirac}(-1 + \alpha) \alpha + 2 - 2 x_0 + 2 x_0 \alpha) \right\}$$

$$\left\{ \begin{array}{ll} 0 & \alpha = -1 \\ \sin(t) t + \frac{1}{2} \cos(t) & \alpha = 1 \\ -\frac{\cos(\alpha t)}{-1 + \alpha} & \text{otherwise} \end{array} \right\}, y(t) = \cos(t) y_0 + \frac{1}{2} \frac{1}{-1 + \alpha} (\sin(t) ($$

$$-2 \operatorname{Dirac}(\alpha + 1) - \operatorname{Dirac}(-1 + \alpha) + 2 \operatorname{Dirac}(-1 + \alpha) \operatorname{Dirac}(\alpha + 1) - \operatorname{Dirac}(-1$$

$$+ \alpha) \alpha + 2 - 2 x_0 + 2 x_0 \alpha) - \left\{ \begin{array}{ll} 0 & \alpha = -1 \\ \cos(t) t + \frac{1}{2} \sin(t) & \alpha = 1 \\ \frac{\alpha \sin(\alpha t)}{-1 + \alpha} & \text{otherwise} \end{array} \right\} + \sin(\alpha t) \left. \right\}$$

L'énoncé de l'exercice demande d'écarter  $\alpha = -1$  et  $\alpha = 1$ . Aussi on prendra  $\alpha = 2$  pour construire les graphiques.

```
> alpha:=2;
soleqd;#récupération de la solution
assign(soleqd);#assignation de la solution
X:=unapply(x(t),(x0,y0,t));#passage à une fonction de trois
variables
Y:=unapply(y(t),(x0,y0,t));#passage à une fonction de trois
variables
```

$$\alpha := 2$$

$$\left\{ x(t) = -\sin(t) y_0 + \frac{1}{2} \cos(t) (2 + 2 x_0) - \cos(2 t), y(t) = \cos(t) y_0 + \frac{1}{2} \sin(t) (2 + 2 x_0) - \sin(2 t) \right\}$$

$$X := (x_0, y_0, t) \rightarrow -\sin(t) y_0 + \frac{1}{2} \cos(t) (2 + 2 x_0) - \cos(2 t)$$



$$Y := (x_0, y_0, t) \rightarrow \cos(t) y_0 + \frac{1}{2} \sin(t) (2 + 2 x_0) - \sin(2 t)$$


(8.4)

On en déduit les trajectoires pour les conditions initiales données dans l'énoncé.

```
> plot([X(0,0,t),Y(0,0,t)],t=0..4*Pi);
plot([X(0,2,t),Y(0,2,t)],t=0..4*Pi);
plot([X(0,4,t),Y(0,4,t)],t=0..4*Pi);
```

```
plot([X(0,6,t),Y(0,6,t)],t=0..4*Pi);
```






Vue l'expression des solutions, les orbites sont évidemment périodiques comme le montre la méthode de construction à partir des solutions explicites :

```
> g1:=plot([X(0,0,t),Y(0,0,t),t=0..2*Pi],x=-6..8,y=-7..7,color=
black);
g2:=plot([X(0,2,t),Y(0,2,t),t=0..2*Pi],x=-6..8,y=-7..7,color=
blue);
g3:=plot([X(0,4,t),Y(0,4,t),t=0..2*Pi],x=-6..8,y=-7..7,color=
red);
g4:=plot([X(0,6,t),Y(0,6,t),t=0..2*Pi],x=-6..8,y=-7..7,color=
green);
with(plots):
display(g1,g2,g3,g4);
g1 := PLOT(...)
g2 := PLOT(...)
g3 := PLOT(...)
```

*g4 := PLOT(...)*




ou la méthode faisant appel à la commande **DEplot**.

```
> restart;  
eqd:={diff(x(t),t)=-y(t)+sin(2*t),diff(y(t),t)=x(t)-cos(2*t)}  
;  
inc:={x(t),y(t)};  
with(DEtools):  
DEplot(eqd,inc,t=0..2*Pi,[[x(0)=0,y(0)=0],[x(0)=0,y(0)=2],[x  
(0)=0,y(0)=4],[x(0)=0,y(0)=6]],stepsize=0.01,linecolor=  
[black,blue,red,green],thickness=1);
```

$$eqd := \left\{ \frac{d}{dt} x(t) = -y(t) + \sin(2t), \frac{d}{dt} y(t) = x(t) - \cos(2t) \right\}$$

$$inc := \{x(t), y(t)\}$$



## ▼ Problème E1 (modèle monétariste inflation-chômage)

### Enoncé

Le modèle monétariste canonique d'inflation chômage repose sur 4 équations structurelles, dont deux sont statiques et deux sont dynamiques.

La première équation statique décrit la formation des salaires nominaux à la manière de M. Friedman réinterprétant la fameuse relation de Phillips : le taux de croissance des salaires nominaux, noté  $w$ , dépend négativement du niveau du taux de chômage, noté  $U$ , et positivement du taux d'inflation anticipé, noté  $\pi$ . Plus le taux de chômage est élevé, moins les salariés sont revendicatifs et "gourmands". Plus les anticipations d'inflation sont élevées, plus ils demandent des augmentations salariales afin de conserver leur pouvoir d'achat. Sous sa forme linéaire, l'équation s'écrit :

$$(1) \quad w(t) = -\beta U(t) + \alpha + h \pi(t) \text{ avec } \alpha \text{ et } \beta \text{ réels strictement positifs et } 0 < h \leq 1$$

La deuxième équation statique dérive le taux d'inflation effectif, noté  $p$ , du taux de croissance des salaires nominaux. Comme les firmes fixent leur prix suivant le principe du mark-up sur les coûts et que les coûts sont essentiellement les coûts salariaux, le taux de croissance des prix est la différence entre le taux de croissance des salaires nominaux et le taux de croissance de la productivité du travail supposé constant, noté  $T$ . On a donc :

$$(2) \quad p(t) = w(t) - T$$

La première équation dynamique explique la formation des anticipations de prix par les agents économiques. L'hypothèse d'anticipations adaptatives stipule que les variations d'anticipations de prix corrigent l'écart constaté entre l'inflation effective et l'inflation prévue :

$$(3) \quad \pi'(t) = j (p(t) - \pi(t)) \text{ avec } 0 < j \leq 1$$

La deuxième équation dynamique décrit l'effet de l'inflation sur les variations du taux de chômage. Pour la comprendre, il faut se placer dans une optique monétariste de la macroéconomie. Les agents dépensent selon l'état de leurs encaisses réelles et donc les variations de la demande globale dépendent des variations de leurs encaisses réelles, soit  $m - p$ . La banque centrale contrôle  $m$ , le taux de croissance de la masse monétaire, qu'on suppose constant. Ainsi, toute augmentation des encaisses réelles a un effet stimulant sur l'activité et fait diminuer le taux de chômage et, inversement, toute baisse des encaisses réelles déprime la demande et fait monter le taux de chômage. Cette relation de causalité des encaisses vers les variations du chômage s'écrit :

$$(4) \quad U'(t) = -k (m - p(t)) \text{ avec } k > 0$$

L'objectif de cet exercice est d'étudier les évolutions conjointes du taux de chômage et des

anticipations de prix.

1. A l'aide des équation (1) et (2), donner la relation entre le taux d'inflation et le taux de chômage.
2. En déduire que l'évolution conjointe des anticipations de prix et du taux de chômage est décrite par un système différentiel linéaire.
3. Ecrire ce système différentiel sous la forme matricielle  $X' = A X + B$ . Montrer que la matrice  $A$  est différentiellement stable.
4. Déterminer l'équilibre dynamique du système différentiel. En tenant compte de la question précédente, conclure sur sa nature.
5. On veut représenter plusieurs trajectoires pour les solutions et les orbites correspondantes dans le plan  $\pi_0 U$ . Pour cela, on attribue les valeurs suivantes aux paramètres :  
 $\alpha = \frac{1}{2}$ ,  $T = \frac{1}{6}$ ,  $m = 10$ ,  $h = 1$ ,  $j = \frac{3}{4}$  et  $k = \frac{1}{2}$  et on fait varier le paramètre  $\beta$  pour repérer trois types de convergence des solutions vers l'équilibre asymptotique. Quels sont ces trois types de convergence. Illustrer chacun des cas par un jeu de trajectoires et des orbites.

### Solution

Pour préparer l'étude, commençons par écrire les équations du modèle comme on le ferait "à la main". Détail qui a son importance : on se gardera bien d'utiliser la variable **Pi** pour désigner le taux d'inflation anticipé car c'est un "mot réservé" et on notera ce dernier **pi**.

```
> restart;  
eq1:=w=-beta*U+alpha+h*pi;  
eq2:=p=w-T;  
eq3:=diff(pi(t),t)=j*(p(t)-pi(t));  
eq4:=diff(U(t),t)=-k*(m-p(t));
```

$$\begin{aligned} eq1 &:= w = -\beta U + \alpha + h \pi \\ eq2 &:= p = w - T \\ eq3 &:= \frac{d}{dt} \pi(t) = j (p(t) - \pi(t)) \\ eq4 &:= \frac{d}{dt} U(t) = -k (m - p(t)) \end{aligned} \quad (9.1)$$

1. On se sert des propriétés de **assign** appliquée à **eq1** : l'égalité est transformée en assignation de **w**. Il suffit alors de demander **eq2** pour obtenir la relation demandée sous forme d'une équation.

```
> assign(eq1);  
eq2;
```

$$p = -\beta U + \alpha + h \pi - T \quad (9.2)$$

2. A partir de cette question, on entre dans la dynamique et il convient de remplacer les variables endogènes par des fonctions-expressions du temps. L'équation **eq2** doit être réécrite à l'aide de **subs**.

```
> eq2:=subs({p=p(t),U=U(t),pi=pi(t)},eq2);
```

$$eq2 := p(t) = -\beta U(t) + \alpha + h \pi(t) - T \quad (9.3)$$

On en tire l'assignation de **p(t)** :

```
> assign(eq2);
```

Alors, **eq3** et **eq4** se présentent automatiquement comme deux équations différentielles linéaires du premier ordre à coefficients constants et avec second membre constant.

```
> eq3;
eq4;
```

$$\begin{aligned} \frac{d}{dt} \pi(t) &= j (-\beta U(t) + \alpha + h \pi(t) - T - \pi(t)) \\ \frac{d}{dt} U(t) &= -k (m + \beta U(t) - \alpha - h \pi(t) + T) \end{aligned} \quad (9.4)$$

Les manipulations suivantes rendent plus lisibles ces deux équations.

```
> eq3:=lhs(eq3)=collect(expand(rhs(eq3)),[pi(t),U(t),j]);
eq4:=lhs(eq4)=collect(expand(rhs(eq4)),[pi(t),U(t),k]);
```

$$\begin{aligned} eq3 &:= \frac{d}{dt} \pi(t) = (h-1)j\pi(t) - j\beta U(t) + (\alpha - T)j \\ eq4 &:= \frac{d}{dt} U(t) = kh\pi(t) - k\beta U(t) + (-m + \alpha - T)k \end{aligned} \quad (9.5)$$

En conclusion, la loi d'évolution conjointe de  $\pi$  et  $U$  est donnée par le système différentiel :

```
> sysdif:=eq3,eq4;
```

$$\begin{aligned} sysdif := \frac{d}{dt} \pi(t) &= (h-1)j\pi(t) - j\beta U(t) + (\alpha - T)j, \frac{d}{dt} U(t) = kh\pi(t) - k\beta U(t) \\ &+ (-m + \alpha - T)k \end{aligned} \quad (9.6)$$

3. Le passage sous forme matricielle du système différentiel **sysdif** se fait automatiquement avec la commande **GenerateMatrix** du paquetage **LinearAlgebra**. Pour montrer que  $A$  est d-stable, on a besoin de connaître les signes de son déterminant et de sa trace, d'où les tests effectués avec **signum** sous les conditions que pose le modèle sur les paramètres.

```
> with(LinearAlgebra):
A,B:=GenerateMatrix([rhs(eq3)=0,rhs(eq4)=0],[pi(t),U(t)]);
detA:=Determinant(A);
trA:=Trace(A);
signum(detA) assuming 0<h,h<=1,0<j,j<=1,k>0,beta>0;#test sur
le signe du déterminant de A
signum(trA) assuming 0<h,h<=1,0<j,j<=1,k>0,beta>0;#test sur
le signe de la trace de A
```

$$\begin{aligned} A, B &:= \begin{bmatrix} (h-1)j & -j\beta \\ kh & -k\beta \end{bmatrix}, \begin{bmatrix} -(\alpha - T)j \\ -(-m + \alpha - T)k \end{bmatrix} \\ detA &:= jk\beta \\ trA &:= (h-1)j - k\beta \\ &1 \\ &-1 \end{aligned} \quad (9.7)$$

Ainsi le déterminant de  $A$  est strictement positif et sa trace strictement négative. Les valeurs propres du polynôme caractéristique sont de partie réelle négative et par conséquent  $A$  est d-stable. Toutes les trajectoires-solutions sont convergentes.

4. L'équilibre dynamique s'obtient avec **LinearSolve** :

```
> Estar:=LinearSolve(A,B);
```

$$Estar := \begin{bmatrix} m \\ -\frac{-hm + T + m - \alpha}{\beta} \end{bmatrix} \quad (9.8)$$

Comme  $A$  est d-stable, l'équilibre est globalement stable.

5. Les valeurs propres de  $A$  sont des réels distincts, confondues ou des complexes conjugués suivant que le discriminant du polynôme caractéristique est strictement positif, nul ou strictement négatif.

Commençons par exprimer ce discriminant qui est égal à  $trA^2 - 4 detA$ .

> **disc:=trA^2-4\*detA;**

$$disc := ((h-1)j - k\beta)^2 - 4jk\beta \quad (9.9)$$

L'énoncé suggère de faire varier le paramètre  $\beta$  pour des valeurs fixées des autres paramètres. La commande **solve**, qui s'applique à des égalités ou des inégalités, permet de dégager 3 cas de figure.

> **alpha,T,m,h,j,k:=1/2,1/6,10,1,3/4,1/2;#paramétrage  
quantitatif  
disc;#récupération de disc  
solve({disc=0,beta>0},beta);#cas d'une valeur propre double  
solve({disc<0,beta>0},beta);#cas de valeurs propres complexes  
conjuguées  
solve({disc>0,beta>0},beta);#cas de valeurs propres réelles**

$$\alpha, T, m, h, j, k := \frac{1}{2}, \frac{1}{6}, 10, 1, \frac{3}{4}, \frac{1}{2}$$

$$\frac{1}{4} \beta^2 - \frac{3}{2} \beta$$

$$\{\beta = 6\}$$

$$\{0 < \beta, \beta < 6\}$$

$$\{6 < \beta\}$$

(9.10)

Pour tracer les trajectoires et les orbites correspondant à chacun de ces cas, on prendra par exemple :  $\beta = 4$  pour représenter le cas où les deux valeurs propres sont complexes;  $\beta = 6$  pour le cas d'une valeur réelle double;  $\beta = 8$  pour le cas où les deux valeurs propres sont réelles. On étudiera donc successivement les systèmes suivants :

> **beta:=4;sysdif1:=sysdif;  
beta:=6;sysdif2:=sysdif;  
beta:=8;sysdif3:=sysdif;**

$$\beta := 4$$

$$sysdif1 := \frac{d}{dt} \pi(t) = \frac{1}{4} - 3 U(t), \quad \frac{d}{dt} U(t) = \frac{1}{2} \pi(t) - 2 U(t) - \frac{29}{6}$$

$$\beta := 6$$

$$sysdif2 := \frac{d}{dt} \pi(t) = \frac{1}{4} - \frac{9}{2} U(t), \quad \frac{d}{dt} U(t) = \frac{1}{2} \pi(t) - 3 U(t) - \frac{29}{6}$$


$$\beta := 8$$

$$\text{sysdif3} := \frac{d}{dt} \pi(t) = \frac{1}{4} - 6 U(t), \quad \frac{d}{dt} U(t) = \frac{1}{2} \pi(t) - 4 U(t) - \frac{29}{6} \quad (9.11)$$

### Cas $\beta=4$

Le portrait des phases, construit avec **DEPlot**, montre clairement que l'équilibre est un foyer attractif.

```
> restart;
sysdif1 := diff(pi(t), t) = 1/4-3*U(t), diff(U(t), t) = (1/2)
*pi(t)-2*U(t)-29/6;
with(DEtools):
DEplot([sysdif1],[pi(t),U(t)],t=0..10,[[pi(0)=8,U(0)=0.05],
[pi(0)=9,U(0)=0.1],[pi(0)=11,U(0)=0.15],[pi(0)=12,U(0)=0.1]],
stepsize=0.01,linecolor=[black,red,blue,green],thickness=2);
sysdif1 := \frac{d}{dt} \pi(t) = \frac{1}{4} - 3 U(t), \frac{d}{dt} U(t) = \frac{1}{2} \pi(t) - 2 U(t) - \frac{29}{6}
```



Les trajectoires de  $\pi$  et  $U$  convergent vers l'équilibre avec des oscillations amorties. Pour représenter ce phénomène, on commence par extraire les solutions obtenues avec **dsolve** en les considérant comme des fonctions de 3 variables : les deux conditions initiales et le temps :

```
> sol1:=dsolve({sysdif1,pi(0)=pi0,U(0)=U0},{pi(t),U(t)}):
assign(sol1);
solpi1:=unapply(pi(t),(pi0,U0,t));
solU1:=unapply(U(t),(pi0,U0,t));
```



$$\text{solpi1} := (\pi_0, U_0, t) \rightarrow 10 + e^{-t} \left( -\frac{1}{4} \sin\left(\frac{1}{2} \sqrt{2} t\right) (39 - 4 \pi_0 + 12 U_0) \sqrt{2} \right. \\ \left. + \cos\left(\frac{1}{2} \sqrt{2} t\right) (-10 + \pi_0) \right)$$

$$\text{solU1} := (\pi_0, U_0, t) \rightarrow \frac{1}{12} + \frac{1}{6} e^{-t} \left( -\frac{1}{2} \sin\left(\frac{1}{2} \sqrt{2} t\right) (39 - 4 \pi_0 + 12 U_0) \sqrt{2} \right. \\ \left. + \frac{1}{2} \cos\left(\frac{1}{2} \sqrt{2} t\right) (39 - 4 \pi_0 + 12 U_0) + 2 \cos\left(\frac{1}{2} \sqrt{2} t\right) (-10 + \pi_0) \right) \quad (9.12)$$

$$+ \sin\left(\frac{1}{2} \sqrt{2} t\right) \sqrt{2} (-10 + \pi 0)$$

puis on trace les courbes en attribuant des valeurs aux couples de conditions initiales.

```
> gsolpi1:=plot([solpi1(8,0.05,t),solpi1(12,0.05,t),solpi1(9,
0.1,t),solpi1(11,0.1,t)],t=0..6,title=typeset("Trajectoires
de ",pi,"(",t,")")):gsolpi1;
gsolU1:=plot([solU1(8,0.05,t),solU1(12,0.05,t),solU1(9,0.1,
t),solU1(11,0.1,t)],t=0..6,title=typeset("Trajectoires de ",
U,"(",t,")")):gsolU1;
```




### Cas $\beta=6$

Le portrait des phases, construit avec **DEPlot**, montre clairement que l'équilibre est encore un foyer attractif avec cette particularité que la valeur propre double n'a qu'une seule direction propre.

```
> restart;
sysdif2:=diff(pi(t), t) = 1/4-(9/2)*U(t), diff(U(t), t) =
```

```
(1/2)*pi(t)-3*U(t)-29/6;
with(DEtools):
DEplot([sysdif2],[pi(t),U(t)],t=0..10,[[pi(0)=8,U(0)=0.05],
[pi(0)=9,U(0)=0.1],[pi(0)=11,U(0)=0.15],[pi(0)=12,U(0)=0.1]],
stepsize=0.01,linecolor=[black,red,blue,green],thickness=2);
```

$$\text{sysdif2} := \frac{d}{dt} \pi(t) = \frac{1}{4} - \frac{9}{2} U(t), \quad \frac{d}{dt} U(t) = \frac{1}{2} \pi(t) - 3 U(t) - \frac{29}{6}$$



Les trajectoires de  $\pi$  et  $U$  convergent vers l'équilibre de façon monotone après que l'exponentielle "l'emporte" sur le polynôme en  $t$ . Comme dans le cas précédent, on commence par extraire les solutions obtenues avec **dsolve** en les considérant comme des fonctions de 3 variables : les deux conditions initiales et le temps :

```
> sol2:=dsolve({sysdif2,pi(0)=pi0,U(0)=U0},{pi(t),U(t)});
assign(sol2);
solpi2:=unapply(pi(t),(pi0,U0,t));
solU2:=unapply(U(t),(pi0,U0,t));
```

$$\text{sol2} := \left\{ U(t) = \frac{1}{18} + \frac{1}{9} e^{-\frac{3}{2}t} \left( -\frac{1}{2} + 3t \left( -\frac{59}{4} + \frac{3}{2} \pi_0 - \frac{9}{2} U_0 \right) + 9 U_0 \right), \pi(t) \right. \\ \left. = 10 + e^{-\frac{3}{2}t} \left( -10 + \pi_0 + t \left( -\frac{59}{4} + \frac{3}{2} \pi_0 - \frac{9}{2} U_0 \right) \right) \right\}$$

$$\text{solpi2} := (\pi_0, U_0, t) \rightarrow 10 + e^{-\frac{3}{2}t} \left( -10 + \pi_0 + t \left( -\frac{59}{4} + \frac{3}{2} \pi_0 - \frac{9}{2} U_0 \right) \right)$$

$$\text{solU2} := (\pi_0, U_0, t) \rightarrow \frac{1}{18} + \frac{1}{9} e^{-\frac{3}{2}t} \left( -\frac{1}{2} + 3t \left( -\frac{59}{4} + \frac{3}{2} \pi_0 - \frac{9}{2} U_0 \right) + 9 U_0 \right) \quad (9.13)$$



puis on trace les courbes en attribuant des valeurs aux couples de conditions initiales.

```
> gsolpi2:=plot([solpi2(8,0.05,t),solpi2(12,0.05,t),solpi2(9,
0.1,t),solpi2(11,0.1,t)],t=0..6,title=typeset("Trajectoires
```

```

de ",pi, "(" ,t, ")" ):gsolpi2;
gsolU2:=plot([solU2(8,0.05,t),solU2(12,0.05,t),solU2(9,0.1,
t),solU2(11,0.1,t)],t=0..6,title=typeset("Trajectoires de ",
U,"(",t,")")):gsolU2;

```



### Cas $\beta=8$

Le portrait des phases, construit avec **DEPlot**, montre que l'équilibre est encore un foyer attractif ou puits.


```

> restart;
sysdif3:=diff(pi(t), t) = 1/4-6*U(t), diff(U(t), t) = (1/2)*
pi(t)-4*U(t)-29/6;
with(DEtools):
DEplot([sysdif3],[pi(t),U(t)],t=0..10,[[pi(0)=8,U(0)=0.05],

```

```
[pi(0)=9,U(0)=0.1],[pi(0)=11,U(0)=0.15],[pi(0)=11,U(0)=0.05]
],stepsize=0.01,linecolor=[black,red,blue,green],thickness=2)
;
```

$$\text{sysdif3} := \frac{d}{dt} \pi(t) = \frac{1}{4} - 6 U(t), \quad \frac{d}{dt} U(t) = \frac{1}{2} \pi(t) - 4 U(t) - \frac{29}{6}$$



On procède comme dans les deux cas précédents pour obtenir les trajectoires de  $\pi$  et  $U$ . On commence par extraire les solutions obtenues avec **dsolve** en les considérant comme des fonctions de 3 variables : les deux conditions initiales et le temps :

```
> sol3:=dsolve({sysdif3,pi(0)=pi0,U(0)=U0},{pi(t),U(t)});
assign(sol3);
solpi3:=unapply(pi(t),(pi0,U0,t));
solU3:=unapply(U(t),(pi0,U0,t));
```

$$\text{sol3} := \left\{ U(t) = \frac{1}{6} e^{-t} \left( -\frac{119}{8} + \frac{3}{2} \pi_0 - 3 U_0 \right) + \frac{1}{2} e^{-3t} \left( \frac{39}{8} - \frac{1}{2} \pi_0 + 3 U_0 \right) \right. \\ \left. + \frac{1}{24}, \pi(t) = e^{-t} \left( -\frac{119}{8} + \frac{3}{2} \pi_0 - 3 U_0 \right) + e^{-3t} \left( \frac{39}{8} - \frac{1}{2} \pi_0 + 3 U_0 \right) + 10 \right\}$$



$$\text{solpi3} := (\pi_0, U_0, t) \rightarrow e^{-t} \left( -\frac{119}{8} + \frac{3}{2} \pi_0 - 3 U_0 \right) + e^{-3t} \left( \frac{39}{8} - \frac{1}{2} \pi_0 + 3 U_0 \right) + 10$$

$$\text{solU3} := (\pi_0, U_0, t) \rightarrow \frac{1}{6} e^{-t} \left( -\frac{119}{8} + \frac{3}{2} \pi_0 - 3 U_0 \right) + \frac{1}{2} e^{-3t} \left( \frac{39}{8} - \frac{1}{2} \pi_0 \right. \\ \left. + 3 U_0 \right) + \frac{1}{24} \quad (9.14)$$

puis on trace les courbes en attribuant des valeurs aux couples de conditions initiales.

```
> gsolpi3:=plot([solpi3(8,0.05,t),solpi3(12,0.05,t),solpi3(9,
0.1,t),solpi3(11,0.1,t)],t=0..6,title=typeset("Trajectoires
de ",pi,("(t)"))):gsolpi3;
gsolU3:=plot([solU3(8,0.05,t),solU3(12,0.05,t),solU3(9,0.1,
t),solU3(11,0.1,t)],t=0..6,title=typeset("Trajectoires de ",
```

$U, (" , t, ") ) : gsolU3;$



## ▼ Problème E2 (modèle de Tobin)

### Enoncé

Le modèle de James Tobin abordé ici est fondamentalement une extension du modèle de croissance néo-classique de Solow au sens où la dynamique monétaire interfère sur la dynamique d'accumulation du capital. Le résultat essentiel, présenté en 1965 dans la revue *Econometrica* sous le titre *Money and Economic Growth*, est que le couplage des marchés des produits et de la monnaie rend la croissance instable sous des hypothèses raisonnables.

La toile de fond est une économie fermée où la population active  $L$  croît au taux "naturel" constant

$$n : \frac{L'(t)}{L(t)} = n.$$

Sur le marché des produits, l'équilibre se définit par l'égalité de l'offre et de la demande. En notant  $Q$  la valeur ajoutée,  $C$  la consommation et  $I$  l'investissement brut, on a l'égalité usuelle en macroéconomie :  $Q(t) = C(t) + I(t)$ . L'investissement brut est la somme de l'investissement de remplacement  $\delta K(t)$  qui est le produit du stock de capital  $K$  et du taux de dépréciation constant  $\delta$  ( $\delta \geq 0$ ) et de l'investissement net, égal à la variation instantanée du stock de capital productif, soit  $\frac{d}{dt} K(t)$ . L'offre  $Q$  est donnée par la fonction de production à deux facteurs  $F$  supposée "well behaved" - en particulier, homogène de degré 1 :  $Q(t) = F(K(t), L(t))$ . La consommation des ménages est une fraction constante  $(1 - s)$  de leur budget  $R$ , somme de leurs revenus tirés de l'activité de production qui sont égaux, en économie fermée, à la valeur ajoutée, et de la variation de leurs encaisses réelles  $\frac{d}{dt} \left( \frac{M(t)}{P(t)} \right)$  en notant  $M$  les encaisses nominales et  $P$  le niveau général des prix. Dans ce cadre,  $R(t) = Q(t) + \frac{d}{dt} \left( \frac{M(t)}{P(t)} \right) = F(K(t), L(t)) + \frac{d}{dt} \left( \frac{M(t)}{P(t)} \right)$  et la consommation vaut  $C(t) = (1 - s) \left[ F(K(t), L(t)) + \frac{d}{dt} \left( \frac{M(t)}{P(t)} \right) \right]$ , où le paramètre  $s$  s'interprète évidemment comme la propension constante à épargner et vérifie par conséquent  $0 \leq s < 1$ . En rassemblant tous ces éléments, l'équilibre sur le marché des produits s'écrit :

$$F(K(t), L(t)) = (1 - s) \left[ F(K(t), L(t)) + \frac{d}{dt} \left( \frac{M(t)}{P(t)} \right) \right] + \frac{d}{dt} K(t) + \delta K(t) \quad (1)$$

Passons au marché de la monnaie, dont l'équilibre se fait à l'égalité de l'offre de monnaie  $M_s$  et de la demande de monnaie par les agents  $M_d$ . On suppose que les autorités monétaires créent de la monnaie au taux positif constant  $\theta$  :  $\frac{M_s'(t)}{M_s(t)} = \theta$ . La demande de monnaie est conçue comme une demande d'encaisses réelles par travailleur qui s'explique par un motif de transaction et un motif de spéculation. On pose  $\frac{M_d(t)}{P(t) L(t)} = a \frac{K(t)}{L(t)} - b \frac{P'(t)}{P(t)}$  ce qui signifie que toute élévation du revenu d'activité, indirectement mesuré par l'élévation du stock de capital par tête, booste la demande d'encaisses réelles (la sensibilité de la demande d'encaisses au capital par tête  $a$  est donc strictement positive) alors que l'inflation décourage la constitution d'encaisses (le paramètre  $b$  est donc strictement positif). Il y a équilibre économique et équilibre dynamique sur le marché de la monnaie si on a à la fois  $M_s(t) = M_d(t) = M(t)$  et  $\frac{M'(t)}{M(t)} = \theta$ , soit encore :

$$\frac{M(t)}{P(t) L(t)} = a \frac{K(t)}{L(t)} - b \frac{P'(t)}{P(t)} \quad (2)$$

$$\frac{M'(t)}{M(t)} = \theta \quad (3)$$

### Première partie

L'objectif de cette première partie est d'établir un système différentiel à deux variables :  $k(t)$  et  $m(t)$ .

La variable  $k$  désigne le capital par tête :  $k(t) = \frac{K(t)}{L(t)}$ . La variable  $m$  désigne les encaisses réelles par tête :  $m(t) = \frac{M(t)}{P(t) L(t)}$ .

1. On pose  $q = \frac{Q}{L}$ . Donner l'expression de la fonction de production par tête qu'on notera  $f$  à partir de la fonction de production  $F$ .

2. On pose  $p(t) = \frac{P'(t)}{P(t)}$ . Donner une expression simple de  $\frac{m'(t)}{m(t)}$  en fonction  $\theta$ ,  $n$  et  $p(t)$ .

3. Montrer à partir de la relation (2) que  $p$  est une fonction croissante de  $k$  et décroissante de  $m$ .
4. Dédire des questions précédentes les lois d'évolution du capital par tête et des encaisses réelles par tête. On notera les équations différentielles **eqd1** et **eqd2**.

### Deuxième partie

Soit **sysdif** le système différentiel non linéaire formé par **eqd1** et **eqd2**.

1. Montrer que **sysdif** admet trois équilibre dynamique dont un seul a ses composantes sont strictement positives quand  $n < \theta$ .
2. Montrer que l'équilibre à composantes positives est localement instable et qu'il s'agit d'un équilibre de point-selle.
3. Illustrer les propriétés dynamiques de **sysdif** en construisant le portrait des phases pour les valeurs suivantes des paramètres :

$$s = 0.1, \theta = 0.025, n = 0.005, \delta = 0.1, a = 0.1, b = 0.05$$

et en supposant que la fonction de production est la Cobb-Douglas  $f(k) = k^{\frac{1}{3}}$ .

### Troisième partie

On a critiqué la formulation que donne Tobin sur les comportements monétaires. Reprendre l'étude du modèle en supposant à présent que la demande d'encaisses réelles est une fonction croissante du capital par tête et du taux d'inflation (en période d'inflation, les agents sont enclins à reconstituer

leurs encaisses) :  $\frac{Md(t)}{P(t)L(t)} = a \frac{K(t)}{L(t)} + b \frac{P'(t)}{P(t)}$  avec  $a$  et  $b$  strictement positifs.

### Solution

#### Première partie

Dans ce genre de problème, il est indispensable de bien préparer sa feuille de travail en commençant par poser les fonctions qui seront appelées souvent dans le cours de l'étude.

> **restart;**

**KF:=t->K(t);#stock de capital**

**LF:=t->L(t);#stock de travail**

**kF:=KF/LF;#capital par tête**

**QF:=(K,L)->F(K,L);#fonction de production**

**MF:=t->M(t);#masse monétaire**

**PF:=t->P(t);#niveau général des prix**

**mF:=MF/(P\*L);#encaisses réelles par tête**

**pF:=D(PF)/PF;#taux d'inflation**

$$KF := t \rightarrow K(t)$$

$$LF := t \rightarrow L(t)$$

$$kF := \frac{KF}{LF}$$

$$QF := (K, L) \rightarrow F(K, L)$$

$$MF := t \rightarrow M(t)$$

$$PF := t \rightarrow P(t)$$

$$mF := \frac{MF}{P L}$$

$$pF := \frac{D(P)}{P}$$

(10.1)

1. Comme  $F$  est homogène de degré 1, on a  $\frac{1}{L}Q = F\left(\frac{K}{L}, \frac{L}{L}\right)$  d'où  $q = f(k) = F(k, 1)$ .

> `qF:=k->f(k);#fonction de production par tête`  
 $qF := k \rightarrow f(k)$  (10.2)

2. Deux lignes suffisent pour obtenir  $\frac{m'(t)}{m(t)} = \theta - p(t) - n$ .

> `xp1:=expand(D(mF)(t)/mF(t));`  
`xp2:=subs({D(M)(t)/M(t)=theta,D(P)(t)/P(t)=p(t),D(L)(t)/L(t)=n},xp1);`

$$xp1 := \frac{D(M)(t)}{M(t)} - \frac{D(P)(t)}{P(t)} - \frac{D(L)(t)}{L(t)}$$

$$xp2 := \theta - p(t) - n$$
 (10.3)

3. Une fois qu'on a exprimé la variable  $p$  en fonction de  $k$  et  $m$ , on teste le signe des dérivées partielles premières sous les hypothèses faites sur les paramètres. Rappelons que `signum` renvoie -1 si l'expression est strictement négative et 1 si elle est strictement positive.

> `rel2:=m=a*k-b*p;#écriture de la relation 2`  
`rel21:=expand(solve(rel2,p));#résolution et simplification`  
`pf:=unapply(rel21,(m,k));#création d'une fonction de 2 variables`  
`assume(a>0,b>0);#introduction des hypothèses sur les paramètres`  
`signum(diff(pf(m,k),m));#test`  
`signum(diff(pf(m,k),k));#test`  
`a:='a':b:='b':#désassignation obligatoire pour préparer les simulations ultérieures`

$$rel2 := m = a k - b p$$

$$rel21 := -\frac{m}{b} + \frac{a k}{b}$$

$$pf := (m, k) \rightarrow -\frac{m}{b} + \frac{a k}{b}$$

-1  
1

 (10.4)

Le taux d'inflation dépend positivement du capital par tête (mécanisme sous-jacent d'inflation par la demande) et négativement de la croissance des encaisses réelles par tête (ce qui reflète simplement le fait que l'accélération de l'inflation vient baisser les encaisses réelles).

4. Si on cherche des expressions épurées, cette question exige beaucoup de manipulations pour peu de résultats. Il vaut mieux faire une partie des calculs à la main et vérifier leur validité avec Maple.

L'équation monétaire n'est guère difficile à construire.

> `eqd21:=diff(m(t),t)/m(t)=xp2;`  
`eqd22:=subs(p(t)=pf(m(t),k(t)),eqd21);`  
`eqd2:=m(t)*eqd22;`

$$\begin{aligned}
 eqd21 &:= \frac{\frac{d}{dt} m(t)}{m(t)} = \theta - p(t) - n \\
 eqd22 &:= \frac{\frac{d}{dt} m(t)}{m(t)} = \theta + \frac{m(t)}{b} - \frac{a k(t)}{b} - n \\
 eqd2 &:= \frac{d}{dt} m(t) = m(t) \left( \theta + \frac{m(t)}{b} - \frac{a k(t)}{b} - n \right)
 \end{aligned}
 \tag{10.5}$$

En revanche, l'équation d'accumulation du capital par tête est rébarbative :

```

> rell:=QF(KF(t),LF(t))=(1-s)*(QF(KF(t),LF(t))+D(M/P)(t))+D(KF)
(t)+delta*KF(t);#écriture de la relation 1
eqd11:=expand(rell/L(t));#passage à la production par tête

```

$$\begin{aligned}
 rell &:= F(K(t), L(t)) = (1 - s) \left( F(K(t), L(t)) + \frac{D(M)(t)}{P(t)} - \frac{M(t) D(P)(t)}{P(t)^2} \right) \\
 &\quad + D(K)(t) + \delta K(t) \\
 eqd11 &:= \frac{F(K(t), L(t))}{L(t)} = \frac{F(K(t), L(t))}{L(t)} + \frac{D(M)(t)}{P(t) L(t)} - \frac{M(t) D(P)(t)}{P(t)^2 L(t)} \\
 &\quad - \frac{s F(K(t), L(t))}{L(t)} - \frac{s D(M)(t)}{L(t) P(t)} + \frac{s M(t) D(P)(t)}{L(t) P(t)^2} + \frac{D(K)(t)}{L(t)} + \frac{\delta K(t)}{L(t)}
 \end{aligned}
 \tag{10.6}$$

La batterie de tests suivante va permettre de simplifier les étapes de calculs et de simplifications.

```

> is(K(t)/L(t)=kF(t));
is(D(K)(t)/L(t)=D(kF)(t)+kF(t)*D(L)(t)/L(t));
is(M(t)*D(P)(t)/(L(t)*P(t)^2)=mF(t)*pF(t));
is(D(M)(t)/(L(t)*P(t))=mF(t)*(D(M)(t)/M(t)));
true
true
true
true

```

(10.7)

Une série de manipulations, guidée par les tests puis l'usage, amène l'équation différentielle cherchée.

```

> eqd12:=f(k(t))=f(k(t))-s*f(k(t))+(1-s)*(m(t)*diff(M(t),t)/M
(t)-m(t)*diff(P(t),t)/P(t))+diff(k(t),t)+k(t)*diff(L(t),t)/L
(t)+delta*k(t);#réécriture à la main de la relation par tête
eqd13:=isolate(eqd12,diff(k(t),t));#isolement de la dérivée
de k
eqd14:=lhs(eqd13)=collect(rhs(eqd13),[k(t),m(t)]);
#ordonnement
eqd15:=lhs(eqd14)=s*f(k(t))-(n+delta)*k(t)+(1-s)*(p(t)-theta)
*m(t);#réécriture à la main de l'équation
eqd1:=subs(p(t)=pf(m(t),k(t)),eqd15);#réemploi du résultat de
la question 3

```

$$\begin{aligned}
\text{eqd12} &:= f(k(t)) = f(k(t)) - sf(k(t)) + (1-s) \left( \frac{m(t) \left( \frac{d}{dt} M(t) \right)}{M(t)} \right. \\
&\quad \left. - \frac{m(t) \left( \frac{d}{dt} P(t) \right)}{P(t)} \right) + \frac{d}{dt} k(t) + \frac{k(t) \left( \frac{d}{dt} L(t) \right)}{L(t)} + \delta k(t) \\
\text{eqd13} &:= \frac{d}{dt} k(t) = sf(k(t)) - (1-s) \left( \frac{m(t) \left( \frac{d}{dt} M(t) \right)}{M(t)} - \frac{m(t) \left( \frac{d}{dt} P(t) \right)}{P(t)} \right) \\
&\quad - \frac{k(t) \left( \frac{d}{dt} L(t) \right)}{L(t)} - \delta k(t) \\
\text{eqd14} &:= \frac{d}{dt} k(t) = \left( -\frac{\frac{d}{dt} L(t)}{L(t)} - \delta \right) k(t) - (1-s) \left( \frac{\frac{d}{dt} M(t)}{M(t)} - \frac{\frac{d}{dt} P(t)}{P(t)} \right) m(t) \\
&\quad + sf(k(t)) \\
\text{eqd15} &:= \frac{d}{dt} k(t) = sf(k(t)) - (n + \delta) k(t) + (1-s) (p(t) - \theta) m(t) \\
\text{eqd1} &:= \frac{d}{dt} k(t) = sf(k(t)) - (n + \delta) k(t) + (1-s) \left( -\frac{m(t)}{b} + \frac{a k(t)}{b} - \theta \right) m(t) \quad (10.8)
\end{aligned}$$

Le système différentiel **sysdif** rassemble **eqd1** et **eqd2** :

$$\begin{aligned}
&> \text{sysdif} := \{ \text{eqd1}, \text{eqd2} \}; \\
\text{sysdif} &:= \left\{ \frac{d}{dt} k(t) = sf(k(t)) - (n + \delta) k(t) + (1-s) \left( -\frac{m(t)}{b} + \frac{a k(t)}{b} - \theta \right) m(t), \quad (10.9) \right. \\
&\quad \left. \frac{d}{dt} m(t) = m(t) \left( \theta + \frac{m(t)}{b} - \frac{a k(t)}{b} - n \right) \right\}
\end{aligned}$$

## Seconde partie

1. Les équilibres sont les couples de réels  $(k, m)$  correspondant à des fonctions  $k(t)$  et  $m(t)$  constantes, annulant par conséquent les dérivées  $\frac{d}{dt} k(t)$  et  $\frac{d}{dt} m(t)$  du système différentiel. Ils vérifient donc les équations **eqe1** et **eqe2** suivantes.

$$\begin{aligned}
&> \text{eqe1} := \text{subs}(\{m(t)=m, k(t)=k\}, \text{rhs}(\text{eqd1})) = 0; \\
&\quad \text{eqe2} := \text{subs}(\{m(t)=m, k(t)=k\}, \text{rhs}(\text{eqd2})) = 0; \\
\text{eqe1} &:= sf(k) - (n + \delta) k + (1-s) \left( -\frac{m}{b} + \frac{a k}{b} - \theta \right) m = 0 \\
\text{eqe2} &:= m \left( \theta + \frac{m}{b} - \frac{a k}{b} - n \right) = 0 \quad (10.10)
\end{aligned}$$

Commençons par étudier l'équation la plus facile : **eqe2**. Deux expressions de  $m$  conviennent.

$$\begin{aligned}
&> \text{solm} := \text{solve}(\text{eqe2}, m); \\
\text{solm} &:= 0, -\theta b + a k + n b \quad (10.11)
\end{aligned}$$

Isolons chacune des solutions. Le petit programme suivant assure que la solution nulle est assignée à la variable **solm1** et la solution non triviale à **solm2**.

$$\begin{aligned}
&> \text{if solm}[1]=0 \text{ then solm1}:=\text{solm}[1]; \text{solm2}:=\text{solm}[2];
\end{aligned}$$

```
else solm1:=solm[2];solm2:=solm[1];
end if;
```

$$\begin{aligned} \text{solm1} &:= 0 \\ \text{solm2} &:= -\theta b + a k + n b \end{aligned} \quad (10.12)$$

Le cas  $m=0$  correspond à une dynamique de croissance sans interférence de la croissance monétaire. Il est logique de retrouver les équilibres du modèle de Solow pour le capital par tête. Ce qu'on vérifie en reportant **solm1** dans l'équation d'accumulation du capital :

```
> eqe11:=subs(m=solm1,eqe1);
```

$$\text{eqe11} := sf(k) - (n + \delta) k = 0 \quad (10.13)$$

On doit étudier la fonction  $F(k) = sf(k) - (n + \delta) k$  en tenant compte des propriétés de  $f$

```
> Fek:=unapply(lhs(eqe11),k);
D(Fek)(k);#dérivée première
D(D(Fek))(k);#dérivée seconde
```

$$\begin{aligned} \text{Fek} &:= k \rightarrow sf(k) - (n + \delta) k \\ s D(f)(k) &- n - \delta \\ s D^{(2)}(f)(k) & \end{aligned} \quad (10.14)$$

Primo : la stricte concavité de  $f$  entraîne que  $F$  est strictement concave. Secundo : sur  $\mathbb{R}_+$ ,  $F$

démarre en 0, passe par un maximum en  $k^* = f'^{-1}\left(\frac{n + \delta}{s}\right) > 0$ , puis s'annule puisqu'elle tend vers  $-\infty$ . Donc, il existe un point d'équilibre  $E1 = (\text{solk1}, 0)$  en plus de  $E0 = (0, 0)$ .

Le cas  $m = a k + b(n - \theta)$  décrit une situation où les encaisses viennent influencer l'équilibre de long terme. Le report de cette valeur dans l'équation **eq1** amène la condition :

```
> eqe12:=subs(m=solm2,eqe1);
```

$$\begin{aligned} \text{eqe12} &:= sf(k) - (n + \delta) k + (1 - s) \left( -\frac{-\theta b + a k + n b}{b} + \frac{a k}{b} - \theta \right) (-\theta b + a k \\ &+ n b) = 0 \end{aligned} \quad (10.15)$$

On doit maintenant étudier le tableau de variation de la fonction

$$\begin{aligned} G(k) &= sf(k) - (n + \delta) k + (1 - s) \left( -\frac{-\theta b + a k + n b}{b} + \frac{a k}{b} - \theta \right) (-\theta b + a k + n b) \\ &= sf(k) - \left( n + \delta + \frac{a}{b} n (1 - s) \right) k + (1 - s) n (n - \theta) \end{aligned}$$

```
> Fek:=unapply(lhs(eqe12),k);
simplify(D(Fek)(k));#dérivée première
D(D(Fek))(k);#dérivée seconde
f(0):=0;#valeur de f en 0
simplify(Fek(0));#ordonnée à l'origine
```

$$\begin{aligned} \text{Fek} &:= k \rightarrow sf(k) - (n + \delta) k + (1 - s) \left( -\frac{-\theta b + a k + n b}{b} + \frac{a k}{b} - \theta \right) (-\theta b + a k \\ &+ n b) \\ s D(f)(k) &- n - \delta - n a + n a s \end{aligned}$$

$$\begin{aligned}
& sD^{(2)}(f)(k) \\
& f(0) := 0 \\
& (-1+s)nb(-\theta+n)
\end{aligned}
\tag{10.16}$$

Primo :  $G$  est strictement concave. Secundo : la fonction démarre en  $nb(\theta - n)(1 - s) > 0$  dans l'hypothèse  $\theta > n$ , passe par un maximum en  $k^{**} > 0$  tel que  $k^{**} = f^{-1}\left(\frac{n + \delta}{s} + \frac{1 - s}{s}an\right)$ , puis décroît et tend vers  $-\infty$ . Il existe donc un troisième équilibre pour la dynamique pour lequel les taux de croissance du capital par tête et des encaisses réelles par tête sont strictement positifs quand on suppose  $\theta > n$ .

En résumé, on a obtenu deux équilibres pour lesquels le taux de croissance des encaisses réelles par tête est nul :  $E0 = (0, 0)$  et  $E1 = (solk1, 0)$  et un troisième équilibre  $E2 = (solk2, solm2)$  dont les deux composantes sont positives.

Les deux premiers équilibres sont aussi ceux du modèle de Solow, c'est à dire un modèle de croissance sans création de monnaie. Seul le troisième caractérise une dynamique de croissance avec interaction de la monnaie dans l'hypothèse qu'on privilégiera désormais où la croissance de la masse monétaire  $\theta$  est plus élevée que celle de la démographie.

2. Le linéarisé de **sysdif** au voisinage de  $E2 = (solk2, solm2)$  est le système différentiel linéaire ::

$$\begin{cases}
\frac{d}{dt} k(t) = \frac{\partial^2}{\partial k^2} F(solk2, solm2) (k - solk2) + \frac{\partial^2}{\partial m^2} F(solk2, solm2) (m - solm2) \\
\frac{d}{dt} m(t) = \frac{\partial^2}{\partial k^2} G(solk2, solm2) (k - solk2) + \frac{\partial^2}{\partial m^2} G(solk2, solm2) (m - solm2)
\end{cases}$$

Pour déterminer la nature locale de l'équilibre, on va calculer la matrice jacobienne puis explorer ses caractéristiques au point  $E2$ .

```

> F1:=unapply(subs(f(k)=qF(k),lhs(eqe1)),(k,m));
F2:=unapply(lhs(eqe2),(k,m));
with(LinearAlgebra):
Jac:=Matrix(2,2,[[diff((F1)(k,m),k),diff((F1)(k,m),m)],[diff(
(F2)(k,m),k),diff((F2)(k,m),m)]]);#expression générale de la
jacobienne

```

$$F1 := (k, m) \rightarrow sf(k) - (n + \delta)k + (1 - s) \left( -\frac{m}{b} + \frac{ak}{b} - \theta \right) m$$

$$F2 := (k, m) \rightarrow m \left( \theta + \frac{m}{b} - \frac{ak}{b} - n \right)$$

$Jac :=$  (10.17)

$$\begin{bmatrix}
s \left( \frac{d}{dk} f(k) \right) - n - \delta + \frac{(1-s)am}{b}, & -\frac{(1-s)m}{b} + (1-s) \left( -\frac{m}{b} + \frac{ak}{b} - \theta \right) \\
-\frac{ma}{b}, & \theta + \frac{2m}{b} - \frac{ak}{b} - n
\end{bmatrix}$$

Faute de connaître l'expression exacte de  $solk2$ , on se contente de calculer la jacobienne en  $solm2$ . L'analyse de ses caractéristiques sera menée en tenant compte des propriétés vérifiées par  $solk2$ .

```

> Jac[1,1]:=subs(m=solm2,Jac[1,1]);

```

```

Jac[1,2]:=subs(m=solm2,Jac[1,2]);
Jac[2,1]:=subs(m=solm2,Jac[2,1]);
Jac[2,2]:=simplify(subs(m=solm2,Jac[2,2]));
Jac;#calcul de la jacobienne en m=solm2
Determinant(Jac);#calcul du déterminant de la jacobienne en
m=solm2
Trace(Jac);#calcul de la trace de la jacobienne en m=solm2

```

$$\begin{aligned}
 Jac_{1,1} &:= s \left( \frac{d}{dk} f(k) \right) - n - \delta + \frac{(1-s)a(-\theta b + ak + nb)}{b} \\
 Jac_{1,2} &:= -\frac{(1-s)(-\theta b + ak + nb)}{b} + (1-s) \left( -\frac{-\theta b + ak + nb}{b} + \frac{ak}{b} - \theta \right) \\
 Jac_{2,1} &:= -\frac{(-\theta b + ak + nb)a}{b} \\
 Jac_{2,2} &:= \frac{-\theta b + ak + nb}{b} \\
 \left[ \left[ s \left( \frac{d}{dk} f(k) \right) - n - \delta + \frac{(1-s)a(-\theta b + ak + nb)}{b}, -\frac{(1-s)(-\theta b + ak + nb)}{b} \right. \right. \\
 &\quad \left. \left. + (1-s) \left( -\frac{-\theta b + ak + nb}{b} + \frac{ak}{b} - \theta \right) \right], \right. \\
 &\quad \left. \left[ -\frac{(-\theta b + ak + nb)a}{b}, \frac{-\theta b + ak + nb}{b} \right] \right] \\
 &\quad \frac{(-\theta b + ak + nb) \left( s \left( \frac{d}{dk} f(k) \right) - n - \delta - an + as n \right)}{b} \\
 &\quad s \left( \frac{d}{dk} f(k) \right) - n - \delta + \frac{(1-s)a(-\theta b + ak + nb)}{b} + \frac{-\theta b + ak + nb}{b} \tag{10.18}
 \end{aligned}$$

Le déterminant de la jacobienne est égal à  $\frac{solm2}{b} (sf'(solk2) - (n + \delta) + (1-s)an)$ . Comme  $b$  et  $solm2$  sont strictement positifs, son signe est celui de l'expression entre parenthèses, qui est la dérivée de  $G$  en  $solm2$ . Or on a vu que  $G'$  est décroissante à partir de  $k^{**}$  et que  $solk2 > k^{**}$ . Par conséquent,  $sf'(solk2) - (n + \delta) + (1-s)an < 0$ , ce qui entraîne que le déterminant de la jacobienne est strictement négatif. Ses deux valeurs propres sont réelles et de signe opposé. L'équilibre  $E2 = (solk2, solm2)$  est un équilibre instable de point-selle.

3. Il s'agit d'un exercice de simulation qui vient conforter par des représentations graphiques l'étude théorique menée précédemment. On commence par assigner les divers paramètres du modèle.

```

> restart;
s,theta,n,delta,a,b,alpha:=0.1,0.025,0.005,0.1,0.1,0.05,1/3;
s, theta, n, delta, a, b, alpha := 0.1, 0.025, 0.005, 0.1, 0.1, 0.05, 1/3 \tag{10.19}

```

Le plan de travail comprend quatre étapes : détermination des équilibres; recherche des isoclines; tracés du portrait de phase global et des portraits de phase au voisinage des points d'équilibre; portrait de phase du système différentiel linéarisé au voisinage de l'équilibre à composantes positives.

Recherche des équilibres

Pour les paramètres donnés, on obtient 3 équilibres (c'est logique puisque  $\theta > n$ ).

```
> eq1:=s*k^alpha-(n+delta)*k+(1-s)*(-m/b+a*k/b-theta)*m = 0;
eq2:=m*(theta+m/b-a*k/b-n) = 0;
solkm:=solve({eq1,eq2},{k,m});
eq1 := 0.1 k1/3 - 0.105 k + 0.9 (-20.00000000 m + 2.000000000 k - 0.025) m = 0
eq2 := m (0.020 + 20.00000000 m - 2.000000000 k) = 0
solkm := {k=0., m=0.}, {k=0.9294286409, m=0.}, {k=0.9235496022, m
=0.09135496022} (10.20)
```

L'algorithme suivant ordonne les solutions dans l'ordre  $E0 = (solkm1, solkm2) = (0, 0)$ ,  $E1 = (solkm2, solkm3) = (solkm2, 0)$ ,  $E3 = (solkm3, solkm1)$ . Ce tri est nécessaire pour garder l'ordre quelle que soit la session).

```
> if rhs(solkm[1,1])=0 then solkm1:=0;solkm2:=0;
    if rhs(solkm[2,1])<rhs(solkm[3,1]) then solkm2:=rhs(solkm
[3,1]);solkm3:=rhs(solkm[3,2]);solkm1:=rhs(solkm[2,1]);solkm3:=
rhs(solkm[2,2]);
    else solkm2:=rhs(solkm[2,1]);solkm3:=rhs(solkm[2,2]);
solkm1:=rhs(solkm[3,1]);solkm3:=rhs(solkm[3,2]);
    end if;
elif rhs(solkm[2,1])=0 then solkm1:=0;solkm2:=0;
    if rhs(solkm[1,1])<rhs(solkm[3,1]) then solkm2:=rhs(solkm
[3,1]);solkm3:=rhs(solkm[3,2]);solkm1:=rhs(solkm[1,1]);solkm3:=
rhs(solkm[1,2]);
    else solkm2:=rhs(solkm[1,1]);solkm3:=rhs(solkm[1,2]);
solkm1:=rhs(solkm[3,1]);solkm3:=rhs(solkm[3,2]);
    end if;
elif rhs(solkm[3,1])=0 then solkm1:=0;solkm2:=0;
    if rhs(solkm[1,1])<rhs(solkm[2,1]) then solkm2:=rhs(solkm
[2,1]);solkm3:=rhs(solkm[2,2]);solkm1:=rhs(solkm[1,1]);solkm3:=
rhs(solkm[1,2]);
    else solkm2:=rhs(solkm[1,1]);solkm3:=rhs(solkm[1,2]);
solkm1:=rhs(solkm[2,1]);solkm3:=rhs(solkm[2,2]);
    end if;
end if;
solkm1;solkm2;
solkm2;solkm3;
solkm3;solkm1;

0
0
0.9294286409
0.
0.9235496022
```


(10.21)

Isoclines

L'isocline  $m'=0$  est la réunion de deux demi-droites.

```
> isom:=unapply(solve(eq2/m,m),k);#détermination de l'isocline
non triviale
gisom:=plot([0,isom(k)],k=0..1.4,m=0..0.15,thickness=3,color=
[red,red]):gisom;#représentation des deux morceaux de
l'isocline m'=0
```

$$isom := k \rightarrow -0.001000000000 + 0.1000000000 k$$



L'isocline  $k'=0$  doit être mise en évidence dans le quart nord-est du plan  $kOm$ .

```
> solisok:=solve(eq1,m);#recherche des expressions satisfaisant
à k'=0
isok1:=unapply(solisok[1],k);
isok2:=unapply(solisok[2],k);
gisok:=plot([isok1(k),isok2(k)],k=0..1.4,m=0..0.15,thickness=
3,color=[red,red]):gisok;#représentation graphique
```

$$solisok := 0.05000000000 k - 0.0006250000000$$

$$+ 0.0002083333333 \sqrt{57600. k^2 - 1.35840 \cdot 10^5 k + 9. + 1.28000 \cdot 10^5 k^{1/3}},$$

$$0.05000000000 k - 0.0006250000000$$


$$- 0.0002083333333 \sqrt{57600. k^2 - 1.35840 \cdot 10^5 k + 9. + 1.28000 \cdot 10^5 k^{1/3}}$$

$$isok1 := k \rightarrow 0.05000000000 k - 0.0006250000000$$

$$+ 0.0002083333333 \sqrt{57600. k^2 - 1.35840 \cdot 10^5 k + 9. + 1.28000 \cdot 10^5 k^{1/3}}$$


$$isok2 := k \rightarrow 0.05000000000 k - 0.0006250000000$$

$$- 0.0002083333333 \sqrt{57600. k^2 - 1.35840 \cdot 10^5 k + 9. + 1.28000 \cdot 10^5 k^{1/3}}$$



Le graphique `giso` réunit toutes les isoclines.

```
> with(plots):
giso:=display([gisom,gisok]):giso;#placement des isoclines
nulle dans le plan kOm
```



### Portrait de phase global

Le "rapatriement" des équations différentielles `eqd1` et `eqd2` se fait par copier-coller à partir des résultats de la question 4 de la première partie. Toutes deux sont immédiatement évaluées pour les paramètres donnés.

```
> eqd1:=diff(k(t),t)=s*k(t)^alpha-(n+delta)*k(t)+(1-s)*(-m(t)
/b+a*k(t)/b-theta)*m(t);
eqd2:=diff(m(t),t)=m(t)*(theta+m(t)/b-a*k(t)/b-n);
```


$$\begin{aligned}
 \text{eqd1} &:= \frac{d}{dt} k(t) = 0.1 k(t)^{1/3} - 0.105 k(t) + 0.9 (-20.00000000 m(t) + 2.000000000 k(t) \\
 &\quad - 0.025) m(t) \\
 \text{eqd2} &:= \frac{d}{dt} m(t) = m(t) (0.020 + 20.00000000 m(t) - 2.000000000 k(t)) \quad (10.22)
 \end{aligned}$$

Pour tracer le diagramme de phase, on fait appel, une fois n'est pas coutume, à **phaseportrait** (**DEplot** fait tout aussi bien l'affaire).

```

> with(DEtools):
gph1:=phaseportrait([eqd1,eqd2],[k(t),m(t)],t=0..25,[[k(0)=
0.2,m(0)=0.03],[k(0)=0.22,m(0)=0.025],[k(0)=0.65,m(0)=0.07],
[k(0)=0.65,m(0)=0.06],[k(0)=1.2,m(0)=0.118],[k(0)=1.2,m(0)=
0.12]],k(t)=0..1.4,m(t)=0..0.15,stepsize=0.01,linewidth=
black,thickness=1):gph1:#champ directionnel et 6 orbites
display(giso,gph1);#superposition des isoclines et du
portrait des phases

```




Par souci de précision, on examine ce qui se passe localement pour chaque point d'équilibre.  
En  $(0, 0)$ , on a un équilibre instable.

```


> gph2:=DEplot([eqd1,eqd2],[k(t),m(t)],t=0..15,[[k(0)=0.005,m
(0)=0.015],[k(0)=0.005,m(0)=0.006],[k(0)=0.005,m(0)=0.01]],k
(t)=0..0.1,m(t)=0..0.05,stepsize=0.001,linewidth=black):
display({gph2,giso},view=[0..0.1,0..0.05]);

```



L'équilibre  $(sol1, 0)$  est un attracteur comme il fallait s'y attendre puisqu'il correspond au cas "solovien" usuel.


```
> gph3:=DEplot([eqd1,eqd2],[k(t),m(t)],t=0..20,[[k(0)=0.9,m(0)=0.015],[k(0)=0.94,m(0)=0.015],[k(0)=0.98,m(0)=0.015]],k(t)=0.9..1,m(t)=0..0.016,stepsize=0.01,linecolor=black):
display({gph3,giso},view=[0.9..1,0..0.016]);
```



L'équilibre  $(sol2, solm2)$  est instable comme on l'a vu dans le cas général.

```
> gph4:=DEplot([eqd1,eqd2],[k(t),m(t)],t=0..20,[[k(0)=0.89,m(0)=0.0881],[k(0)=0.89,m(0)=0.089],[k(0)=0.955,m(0)=0.094],[k(0)=0.955,m(0)=0.0944]],k(t)=0.88..0.96,m(t)=0.08..0.105,stepsize=0.01,linecolor=black):
```

```
display({gph4,giso},view=[0.88..0.96,0.08..0.105]);
```



#### Portrait de phase après linéarisation

On veut s'assurer que l'équilibre ( $solk2, solm2$ ) correspond à un point-selle de la dynamique. On va donc linéariser le système différentiel au voisinage de cet équilibre. Tout d'abord, on cherche la jacobienne, et, pendant qu'on y est, on demande ses caractéristiques.

```
> eq1;
eq2;
F1:=unapply(lhs(eq1),(k,m));
F2:=unapply(lhs(eq2),(k,m));
with(LinearAlgebra):
Jac:=Matrix(2,2,[[D[1](F1)(solk3,solm3),D[2](F1)(solk3,solm3)
],[D[1](F2)(solk3,solm3),D[2](F2)(solk3,solm3)]]);#matrice
jacobienne
Determinant(Jac);
Trace(Jac);
Eigenvalues(Jac);
```

$$0.1 k^{1/3} - 0.105 k + 0.9 (-20.00000000 m + 2.000000000 k - 0.025) m = 0$$

$$m (0.020 + 20.00000000 m - 2.000000000 k) = 0$$

$$F1 := (k, m) \rightarrow 0.1 k^{1/3} - 0.105 k + 0.9 (-20.00000000 m + 2.000000000 k - 0.025) m$$

$$F2 := (k, m) \rightarrow m (0.020 + 20.00000000 m - 2.000000000 k)$$

$$Jac := \begin{bmatrix} 0.09458730423 & -1.648889284 \\ -0.1827099204 & 1.827099205 \end{bmatrix}$$

$$-0.1284480414$$

$$1.921686509$$

(10.23)

$$\begin{bmatrix} -0.0646653028361179328 + 0. I \\ 1.98635181206611766 + 0. I \end{bmatrix} \quad (10.23)$$


Comme prévu, l'équilibre est instable : le déterminant est négatif et les 2 valeurs propres sont réelles et de signe opposé. Il s'agit bien d'un point selle ce qu'on va illustrer dans un diagramme de phase.

Explicitons le système différentiel linéarisé :

```
> eqdlin1:=D(k)(t)=Jac[1,1]*(k(t)-solkm3)+Jac[1,2]*(m(t)-solm3);
eqdlin2:=D(m)(t)=Jac[2,1]*(k(t)-solkm3)+Jac[2,2]*(m(t)-solm3);
eqdlin1:=D(k)(t)=0.09458730423 k(t) + 0.06327814771 - 1.648889284 m(t)
eqdlin2:=D(m)(t) = -0.1827099204 k(t) + 0.0018270991 + 1.827099205 m(t) (10.24)
```

Le diagramme renvoyé par **DEplot** est décevant ... Même avec des effets de loupe (zooming!) prononcés. En fait, le bras stable de la dynamique se confond presque avec l'isocline  $m'=0$ .

```
> glin1:=DEplot([eqdlin1,eqdlin2],[k(t),m(t)],t=0..25,[[k(0)=
0.9231,m(0)=0.091312],[k(0)=0.9231,m(0)=0.091309],[k(0)=
0.924,m(0)=0.09139],[k(0)=0.924,m(0)=0.091399]],k(t)=0.923.
.0.9245,m(t)=0.09125..0.0915,stepsize=0.01,linecolor=black):
display({glin1,giso},view=[0.923..0.9245,0.09125..0.0915]);
```



### Troisième partie

En modifiant le comportement monétaire des agents, la relation (2) devient :

$$\frac{M(t)}{P(t)L(t)} = a \frac{K(t)}{L(t)} + b \frac{P'(t)}{P(t)}. \text{ Les relations (1) et (3) sont conservées telles quelles.}$$

Formellement, tout se passe comme si le système différentiel **sysdif** restait le même à ceci près que  $b$  est maintenant négatif.

```
> restart;
eqd1:=diff(k(t), t) = s*f(k(t))-(n+delta)*k(t)+(1-s)*(-m(t)
/b+a*k(t)/b-theta)*m(t);
```

```

eqd2:=diff(m(t), t) = m(t)*(theta+m(t)/b-a*k(t)/b-n);
sysdif := [eqd1,eqd2];

```

$$eqd1 := \frac{d}{dt} k(t) = sf(k(t)) - (n + \delta) k(t) + (1 - s) \left( -\frac{m(t)}{b} + \frac{a k(t)}{b} - \theta \right) m(t)$$

$$eqd2 := \frac{d}{dt} m(t) = m(t) \left( \theta + \frac{m(t)}{b} - \frac{a k(t)}{b} - n \right)$$

$$sysdif := \left[ \frac{d}{dt} k(t) = sf(k(t)) - (n + \delta) k(t) + (1 - s) \left( -\frac{m(t)}{b} + \frac{a k(t)}{b} - \theta \right) m(t), \quad (10.25) \right.$$

$$\left. \frac{d}{dt} m(t) = m(t) \left( \theta + \frac{m(t)}{b} - \frac{a k(t)}{b} - n \right) \right]$$

Les équilibres ont été calculés dans la deuxième partie. On s'intéresse à  $E2$ , l'équilibre "positif", dont la nature est encore révélée par les valeurs propres de la jacobienne associée à **sysdif** calculée en ce point.

On a vu que le déterminant de la matrice jacobienne a pour expression

$$\frac{solm2}{b} (sf'(solk2) - (n + \delta) + (1 - s) a n). \text{ Comme } solk2 \text{ vérifie } sf'(solk2) - (n + \delta) + (1 - s) a n < 0$$

et que  $solm2$  est par construction strictement positif, le fait que  $b$  est maintenant négatif entraîne que le déterminant est strictement positif. La matrice jacobienne a deux valeurs propres réelles en  $E2$ .

La trace a pour expression  $s \left( \frac{d}{dk} f(solk2) \right) - n - \delta + \frac{(1 - s) a solm2}{b} + \frac{solm2}{b}$ . Comme

$solm2 > 0, a > 0, 0 \leq s < 1$  et  $b < 0$ , alors  $\frac{(1 - s) a solm2}{b} + \frac{solm2}{b} < 0$ . De plus,  $sf'(solk2) - n - \delta < sf'(solk2) - (n + \delta) + (1 - s) a n < 0$ . La trace est la somme de termes négatifs et

en conséquence les deux valeurs propres sont toutes deux des réels négatifs. L'équilibre  $E2$  est donc stable et c'est un noeud attractif. La nouvelle hypothèse sur les comportements monétaires a radicalement changé la nature de l'équilibre "positif" : l'état stationnaire était instable dans le modèle de Tobin parce que la monnaie était considérée comme un titre par les agents; il est stable quand les agents désirent la monnaie pour ses qualités de transfert de réserve de valeur et sont sensibles au pouvoir d'achat de leurs encaisses monétaires. Le portrait des phases établi avec les données de la deuxième partie mais en modifiant simplement le signe du paramètre  $b$  illustre bien le phénomène de bifurcation.

```


> sysdifnum:=subs({f(k(t))=k(t)^(1/3),s=0.1,theta=0.025,n=
0.005,delta=0.1,a=0.1,b=-0.05},sysdif);
with(DEtools):
DEplot(sysdifnum,[k(t),m(t)],t=0..100,[[k(0)=0.2,m(0)=0.14],
[k(0)=0.22,m(0)=0.01],[k(0)=1.3,m(0)=0.01],[k(0)=1.3,m(0)=
0.14]],k(t)=0..1.4,m(t)=0..0.15,stepsize=0.01,linewidth=
black,thickness=1);#champ directionnel et 4 orbites

```

$$sysdifnum := \left[ \frac{d}{dt} k(t) = 0.1 k(t)^{1/3} - 0.105 k(t) + 0.9 (20.00000000 m(t) \right.$$

$$\left. - 2.000000000 k(t) - 0.025) m(t), \frac{d}{dt} m(t) = m(t) (0.020 - 20.00000000 m(t) \right.$$

$$\left. + 2.000000000 k(t) \right]$$



## ▼ Problème E3 (modèle de la lutte des classes)

### Enoncé

Ce problème s'appuie sur le modèle de croissance cyclique que R.M. Goodwin a présenté en 1965 au 1<sup>o</sup> Congrès mondial de la Société d'économétrie puis publié sous le titre "A Growth Cycle" dans l'ouvrage édité par C.H. Feinstein "Socialism, Capitalism and Economic Growth", London, Cambridge University Press, pp. 54-58. L'idée générale est de fusionner dans une même structure formelle le phénomène de croissance capitaliste et celui de cycles économiques, successions de phases d'expansion et de récession dont le théoricien cherche les causes endogènes. L'inspiration est à la fois keynésienne (pas de prise en compte des prix de biens, production à facteurs complémentaires, mode de formation des salaires, chômage durable, ...) et marxiste en un sens restreint. La société est divisée en deux classes antagonistes qui luttent chacune pour obtenir la plus grosse part possible du revenu national dans un environnement conjoncturel mouvant. Il ne s'agit donc pas d'un modèle ouvertement révolutionnaire mais d'une analyse plutôt convaincante des conséquences économiques de la bipartition tranchée du corps social.

On considère un espace économique capitaliste fermé (au sens macroéconomique). Les capitalistes, propriétaires des moyens de production, perçoivent des profits  $P$ . Les salariés ont pour revenu des salaires  $W$ . Le revenu national  $Y$ , égal au produit intérieur brut en économie fermée, est la somme des profits et des salaires :  $Y = P + W$ . Il en découle que la part des salaires dans le revenu national est  $s = \frac{W}{Y}$  et celle des profits  $p = \frac{P}{Y}$ .

Les salaires sont intégralement dépensés en consommation  $C$  :  $W = C$  et les profits en investissement  $I$  :  $P = I$ . L'investissement correspond à la variation instantanée du stock de capital  $K$  :

$$\frac{d}{dt} K(t) = I(t).$$

La fonction de production est à facteurs complémentaires avec un trend de croissance constant de la productivité par travailleur salarié. En notant  $L$  la population active employée et  $y = \frac{Y}{L}$  la

productivité moyenne du travail, on a :

$y(t) = y_0 e^{g t}$ , où  $g$  est un taux de croissance positif constant et  $y_0$  un paramètre strictement positif. Conformément aux faits stylisés de la croissance, la productivité moyenne du capital reste constante :  $\frac{Y(t)}{K(t)} = c$ , avec  $c$  constante réelle positive.

La formation des salaires obéit à une dynamique spécifiant les conditions dans lesquelles se font les négociations salariales entre les partenaires sociaux. La toile de fond est la situation démographique. On admet que la population active  $L^*$  croît au taux constant  $n$  ( $> 0$ ) :  $L^*(t) = L_0^* e^{n t}$ . Par définition, le nombre de chômeurs  $U$  est la différence entre la population active et la population active occupée :  $U = L^* - L$ , de sorte que le taux de chômage vaut  $u = (L^* - L)/L^*$  et le taux d'emploi  $v = L/L^*$ . Plus le taux de chômage est élevé, moins les salariés sont en situation de force dans les négociations salariales; inversement, plus le taux d'emploi s'élève, plus les salariés sont en mesure d'arracher des augmentations de salaire. Cette relation positive entre le taux de croissance du salaire moyen

$w = \frac{W}{L}$  et le taux d'emploi s'exprime par la relation de Phillips basique :  $\frac{d}{dt} w(t) = a v(t) - b$ ,

avec  $a$  et  $b$  strictement positifs.

1. Dégager du texte de l'énoncé les 16 équations qui constituent la forme structurelle du modèle de Goodwin. Préciser les variables de niveau, les ratios et les paramètres. Expliquer pourquoi les variables et les ratios doivent être datés.
2. Dédurre de la forme structurelle un système différentiel non linéaire en  $s$  et  $v$ . On pourra commencer par exprimer les taux de croissance des ratios puis chercher une expression du taux de croissance de  $s$  en fonction de  $v$  et une expression du taux de croissance de  $v$  en fonction de  $s$ .
3. Déterminer les deux équilibres de ce système puis étudier leur nature locale en tenant compte des relations d'inégalité  $g + b < a$  et  $g + n < c$ .
4. Représenter graphiquement la dynamique de ce modèle dans le plan des phases  $vOs$ . On prendra les valeurs suivantes pour les paramètres :  $a = 5$ ,  $b = 1$ ,  $g = 1$ ,  $c = 3$ ,  $n = 0$ .

## Solution

1. La forme structurelle du modèle comprend 16 équations pour 9 variables de niveau (revenu national  $Y$ , masse salariale  $W$ , profits  $P$ , population active  $La$ , population active occupée  $L$ , population au chômage  $U$ , consommation  $C$ , investissement  $F$ , stock de capital  $K$ ) et 6 ratios (part des salaires dans le revenu national  $s$ , part des profits dans le revenu national  $p$ , productivité du travail  $y$ , taux de salaire moyen  $w$ , taux de chômage  $u$ , taux d'emploi  $v$ ). Son caractère dynamique est affirmé par la présence de deux dérivées par rapport au temps  $t$  : accumulation du capital productif et variation des salaires. En conséquence, toutes les variables doivent être datées. Les paramètres sont strictement positifs : la productivité du capital  $c$ , la productivité du travail à la date 0, le taux de croissance de la productivité du travail  $g$ , la population active à la date 0, le taux de croissance démographique  $n$  et les deux paramètres de la relation linéaire de Phillips.

> **restart;**

**eq1:=Y(t)=P(t)+W(t);#répartition des revenus**

**eq2:=s(t)=W(t)/Y(t);#part des salaires dans le revenu national**

**eq3:=p(t)=P(t)/Y(t);#part des profits dans le revenu national**

**eq5:=C(t)=W(t);#les salaires financent la consommation**

**eq6:=P(t)=F(t);#les profits financent l'investissement**

**eq7:=F(t)=diff(K(t),t);#l'investissement est la variation du capital**

**eq8:=Y(t)/K(t)=c;#la productivité du capital est constante**

```

eq9:=y(t)=Y(t)/L(t);#définition de la productivité du travail
eq10:=y(t)=y(0)*exp(g*t);#croissance de la productivité du
travail à taux constant
eq11:=La(t)=L(0)*exp(n*t);#croissance de la population active
à taux constant
eq12:=U(t)=La(t)-L(t);#définition de la population active
sans emploi
eq13:=u(t)=U(t)/La(t);#taux de chômage
eq14:=v(t)=L(t)/La(t);#taux d'emploi
eq15:=w(t)=W(t)/L(t);#définition du taux de salaire moyen
eq16:=diff(w(t),t)/w(t)=a*v(t)-b;#relation de Phillips

```

$$eq1 := Y(t) = P(t) + W(t)$$

$$eq2 := s(t) = \frac{W(t)}{Y(t)}$$

$$eq3 := p(t) = \frac{P(t)}{Y(t)}$$

$$eq5 := C(t) = W(t)$$

$$eq6 := P(t) = F(t)$$

$$eq7 := F(t) = \frac{d}{dt} K(t)$$

$$eq8 := \frac{Y(t)}{K(t)} = c$$

$$eq9 := y(t) = \frac{Y(t)}{L(t)}$$

$$eq10 := y(t) = y(0) e^{g t}$$

$$eq11 := La(t) = L(0) e^{n t}$$

$$eq12 := U(t) = La(t) - L(t)$$

$$eq13 := u(t) = \frac{U(t)}{La(t)}$$

$$eq14 := v(t) = \frac{L(t)}{La(t)}$$

$$eq15 := w(t) = \frac{W(t)}{L(t)}$$

$$eq16 := \frac{\frac{d}{dt} w(t)}{w(t)} = a v(t) - b \quad (11.1)$$

2. Pour calculer les taux de croissance des ratios, on utilisera la procédure **txcr** qui calcule la dérivée logarithmique des deux membres d'une équation de définition et remplace les expressions du

type  $\frac{\frac{d}{dt} X(t)}{X(t)}$  par le raccourci commode  $gX$ .

```

> txcr:=proc(eq)
  local xp1,xp2;

```

```

xp1:=expand(diff(map(log,eq),t));
xp2:=subs({diff(s(t),t)/s(t)=gs,diff(W(t),t)/W(t)=gW,diff(Y
(t),t)/Y(t)=gY,diff(K(t),t)/K(t)=gK,diff(w(t),t)/w(t)=gw,diff
(v(t),t)/v(t)=gv,diff(L(t),t)/L(t)=gL,diff(y(t),t)/y(t)=gy,
diff(La(t),t)/La(t)=gLa,diff(p(t),t)/p(t)=gp,diff(P(t),t)/P
(t)=gP,diff(u(t),t)/u(t)=gu,diff(U(t),t)/U(t)=gU},xp1);
xp2;
end proc;
txcr := proc(eq)
  local xp1, xp2;
  xp1 := expand(diff(map(log, eq), t));
  xp2 := subs({diff(K(t), t)/K(t) = gK, diff(L(t), t)/L(t) = gL, diff(La(t), t)/La(t)
  = gLa, diff(P(t), t)/P(t) = gP, diff(U(t), t)/U(t) = gU, diff(W(t), t)/W(t) = gW,
  diff(Y(t), t)/Y(t) = gY, diff(p(t), t)/p(t) = gp, diff(s(t), t)/s(t) = gs, diff(u(t), t)
  /u(t) = gu, diff(v(t), t)/v(t) = gv, diff(w(t), t)/w(t) = gw, diff(y(t), t)/y(t) = gy},
  xp1);
  xp2
end proc

```

L'application de cette procédure aux equations **eq2, eq3, eq8, eq9, eq13, eq14, eq15** donne :

```

> eq2a:=txcr(eq2);
eq3a:=txcr(eq3);
eq8a:=txcr(eq8);
eq9a:=txcr(eq9);
eq13a:=txcr(eq13);
eq14a:=txcr(eq14);
eq15a:=txcr(eq15);

```

$$\begin{aligned}
 eq2a &:= gs = gW - gY \\
 eq3a &:= gp = gP - gY \\
 eq8a &:= gY - gK = 0 \\
 eq9a &:= gy = gY - gL \\
 eq13a &:= gu = gU - gLa \\
 eq14a &:= gv = gL - gLa \\
 eq15a &:= gw = gW - gL
 \end{aligned}$$

A partir des équations **eq2a** et **eq15a**, puis en tenant compte des équations **eq9a** et **eq16**, on tire une première équation différentielle en  $s$  :

```

> eq17:=eq15a-eq2a;
eq18:=isolate(algsubs(rhs(eq9a)=lhs(eq9a),eq17),gs);
eq19:=subs({gs=diff(s(t),t)/s(t),gy=g,gw=rhs(eq16)},eq18);
eqd1:=s(t)*eq19;

```

$$\begin{aligned}
 eq17 &:= gw - gs = gY - gL \\
 eq18 &:= gs = -gy + gw
 \end{aligned}$$

$$eq19 := \frac{\frac{d}{dt} s(t)}{s(t)} = -g + a v(t) - b$$

$$eqd1 := \frac{d}{dt} s(t) = s(t) (-g + a v(t) - b) \quad (11.4)$$

A partir des équations **eq14a** et **eq9a** puis en tenant compte de **eq8a**, **eq10**, **eq11**, **eq3**, **eq6**, **eq7** et du fait que  $p = 1 - s$  (la somme des parts des salaires et des profits dans le revenu national est égale à 1), on obtient une équation différentielle en  $v$  :

```
> eq20:=subs({gL=a*n,gy=g},isolate(eq9a+eq14a,gv));
eq21:=subs(isolate(eq8a,gY),eq20);
eq22:=subs(gK=(1-s)*c,eq21);
eq23:=subs({gv=diff(v(t),t)/v(t),s=s(t)},eq22);
eqd2:=v(t)*eq23;
```

$$eq20 := gv = gY - n - g$$

$$eq21 := gv = gK - n - g$$

$$eq22 := gv = (1 - s) c - n - g$$

$$eq23 := \frac{\frac{d}{dt} v(t)}{v(t)} = (1 - s(t)) c - n - g$$

$$eqd2 := \frac{d}{dt} v(t) = v(t) ((1 - s(t)) c - n - g) \quad (11.5)$$

D'où le système différentiel non linéaire en  $s$  et  $v$  :

```
> sysdif:=[eqd1,eqd2];
sysdif:= [ \frac{d}{dt} s(t) = s(t) (-g + a v(t) - b), \frac{d}{dt} v(t) = v(t) ((1 - s(t)) c - n - g) ] \quad (11.6)
```

3. La recherche des équilibres est celle des valeurs de  $s$  et  $v$  annulant les dérivées  $\frac{d}{dt} s(t)$  et  $\frac{d}{dt} v(t)$ .

```
> sysdifeq:=subs({diff(s(t),t)=0,diff(v(t),t)=0,s(t)=s,v(t)=v},
sysdif);
solve(sysdifeq,[s,v]);
equi1:=%[1];
equi2:=%[2];
sysdifeq := [0 = s (-g + a v - b), 0 = v ((1 - s) c - n - g)]
[ [s=0, v=0], [s = \frac{c-n-g}{c}, v = \frac{g+b}{a} ] ]
equi1 := [s=0, v=0]
equi2 := [s = \frac{c-n-g}{c}, v = \frac{g+b}{a} ] \quad (11.7)
```

Grâce aux hypothèses sur les paramètres et l'hypothèse additionnelle  $c > n + g$ , on voit que la solution non triviale a ses deux composantes positives.

Pour étudier la nature locale de chaque équilibre, on linéarise le système différentiel, ce qui

contraint à calculer la matrice jacobienne.

```

> F1:=unapply(rhs(sysdifeq[1]),(s,v));#
F2:=unapply(rhs(sysdifeq[2]),(s,v));
with(LinearAlgebra):
Jac:=Matrix(2,2,[[diff(F1(s,v),s),diff(F1(s,v),v)],[diff(F2
(s,v),s),diff(F2(s,v),v)]]);
      F1 := (s, v) → s (-g + a v - b)
      F2 := (s, v) → v ((1 - s) c - n - g)
      Jac := 
$$\begin{bmatrix} -g + a v - b & s a \\ -v c & (1 - s) c - n - g \end{bmatrix}$$


```

(11.8)

Au point d'équilibre (0,0), la jacobienne vaut :

```

> Jacequi1:=subs(equi1,Jac);
      Jacequi1 := 
$$\begin{bmatrix} -g - b & 0 \\ 0 & c - n - g \end{bmatrix}$$


```

(11.9)

On voit que la matrice est diagonale. Ses valeurs propres sont réelles et de signe opposé puisque  $c > g + n$  et tous les paramètres sont positifs. L'équilibre (0, 0) est instable et c'est un col.

On procède de la même manière au point d'équilibre  $\left(\frac{c-n-g}{c}, \frac{g+b}{a}\right)$ .

```

> Jacequi2:=subs(equi2,Jac);#calcul de la jacobienne
Jacequi2[2,2]:=simplify(Jacequi2[2,2]);
Jacequi2;#jacobienne après simplification
Determinant(Jacequi2);#déterminant
Trace(Jacequi2);#trace
disc:=Trace(Jacequi2)^2-4*Determinant(Jacequi2);#discriminant
du polynôme caractéristique
      Jacequi2 := 
$$\begin{bmatrix} 0 & \frac{(c-n-g)a}{c} \\ -\frac{(g+b)c}{a} & \left(1 - \frac{c-n-g}{c}\right)c - n - g \end{bmatrix}$$

      Jacequi22,2 := 0
      
$$\begin{bmatrix} 0 & \frac{(c-n-g)a}{c} \\ -\frac{(g+b)c}{a} & 0 \end{bmatrix}$$

      (c - n - g) (g + b)
      0
      disc := -4 (c - n - g) (g + b)

```

(11.10)

Le discriminant du polynôme caractéristique est négatif, donc les valeurs propres sont complexes conjuguées. Comme leur somme est nulle, ce sont des imaginaires purs. L'équilibre est un centre : il n'est ni un attracteur ni un répulseur.

4. On commence par récupérer le système différentiel où les paramètres sont assignés.


```
> a,b,g,c,n:=5,1,1.5,3,0;
sysdif;
```

```
a, b, g, c, n := 5, 1, 1.5, 3, 0
```

$$\left[ \frac{d}{dt} s(t) = s(t) (-2.5 + 5 v(t)), \frac{d}{dt} v(t) = v(t) (1.5 - 3 s(t)) \right] \quad (11.11)$$

Le portrait des phases est généré par la commande **DEplot** du paquetage **DEtools** qui permet d'obtenir simultanément le champ directionnel et des orbites.

```
> with(DEtools):
DEplot(sysdif,[s(t),v(t)],t=0..2*Pi,[[s(0)=0.5,v(0)=0.6],[s
(0)=0.5,v(0)=0.75],[s(0)=0.5,v(0)=0.86]],s(t)=0..1,v(t)=0..1,
stepsize=0.01,thickness=1,linecolor=black,method=rkf45,title=
`Diagramme de phases du modèle de Goodwin`);
```



Le graphique confirme l'analyse théorique : l'équilibre non trivial est un centre autour duquel s'enroulent des orbites en courbes fermées. L'interprétation de ce phénomène de croissance accompagnée de cycles endogènes ressort des équations structurelles. Une période d'expansion est marquée par un recul de "l'armée de réserve industrielle". L'augmentation du taux d'emploi place les travailleurs salariés en position de force dans les négociations salariales. Les salaires augmentent. Quand leur accroissement est plus fort que celui de la productivité du travail, la part des salaires dans le revenu national augmente alors que la part des profits se met à diminuer. Les investissements sont automatiquement révisés à la baisse. La production recule et les firmes licencient. La phase de récession est marquée par une diminution continue des rémunérations du travail qui va peu à peu améliorer la part des profits dans le revenu national, donc relancer les investissements, la production puis l'emploi ...